


# 44 Programs for the TRS-80 Model 100 Portable Computer

A practical, useful collection of easy-to-use ready-to-run programs in BASIC for businessmen, teachers, students, professionals and hobbyists using the TRS-80 Model 100 and NEC PC-8200 series of portable computers.



by Jim Cole

**44 Programs  
for the  
TRS-80 Model 100  
Portable Computer**

### **Programming books by Jim Cole**

99 Tips & Tricks for the New Pocket Computers  
Practical PC-2/PC-1500 Pocket Computer Programs  
Pocket Computer Programming Made Easy  
101 Pocket Computer Programming Tips & Tricks  
Murder In The Mansion and Other Computer Adventures  
50 Programs in BASIC for Home, School & Office  
50 MORE Programs in BASIC for Home, School & Office  
35 Practical Programs for the CASIO Pocket Computer  
Pocket Computer Program Writing Workbook

# **44 Programs for the TRS-80 Model 100 Portable Computer**

**by Jim Cole**

**ARCsoft Publishers**  
**Woodsboro, Maryland**



FIRST EDITION  
FIRST PRINTING

© 1984 by ARCsoft Publishers, P.O. Box 132, Woodsboro, MD 21798 USA

Reproduction or publication of the contents of this book, in any manner, without express permission of the publisher, is prohibited. No liability is assumed with respect to the use of the information herein.

Trademark credits and software copyrights:

Radio Shack, TRS-80 and Model 100 are trademarks of Tandy Corporation.

Microsoft BASIC is a trademark of Microsoft Inc.

Applications software, programming advice, and programs in this book are Copyright 1984 by ARCsoft Publishers.

ISBN 0-86668-034-9

# Preface

The microcomputers from Tandy Corporation/Radio Shack are among the most popular around the world for use in the home, the classroom, the business office. In fact, all of the various TRS-80 models, taken together, are the all-time best-selling personal computers to date.

The lightweight lap-sized design of the Model 100 Micro Executive Work Station, its powerful Microsoft BASIC, its built-in programs, its communications capability, all add up to place the Model 100 at the forefront of the new wave of portable computers for use anywhere by students, businessmen, teachers, parents, professionals, hobbyists, scientists, doctors, lawyers, engineers, and all others who want to learn and use the new technology.

This powerful computer is not a toy! It is an executive work station. It is a communications station. It is a teacher, a helper, a powerful tool. The hardware and software combination adds up to one highly useful machine for the business environment, the classroom, and the home.

The total number of applications to which the Model 100 Portable Computer can be put is limited only by the scope of the imagination. In this book, we have attempted to lead the way by creating and sharing 44 new, practical sets of applications software for your use.

This book, as well as all published by *ARCsoft Publishers*, is written for newcomers, novices and first-timers, as well as for advanced users of microcomputers needing new ideas. Our intention has been to provide 44 easy-to-type-in-and-run programs for the TRS-80 Model 100 and other computers using the same Microsoft version of the BASIC computer programming language. You type these programs into your computer and it does the rest. You do not have to be a programmer or program writer to use this book.

— Jim Cole

# Table of Contents

Introduction	8
--------------	---

## **Practical programs for home, school and office**

### *Business interests:*

Making Change	13
Advertising Costs	15
Daily Codes	17
Inventory	19
Daily Receipts Adder	24
Income Property Cash Flow Analysis	26
Executive Decision Maker	28

### *Personal improvements:*

Temperature Converter	28
Jogger Logger	31
Dual-Time Universal Clock	37
Bubble Sort	39
Day of the Year	40
Name 'N Note List	42
Code Breaker	46
Decimal-to-Binary Number Converter	49
Hour Clock	50
Code Practice Oscillator	51
Random Number Tip	55
Alarm Clock	55
Radio Frequency & Wavelength	57

*Personal finance:*

How Money Grows	59
Deposit Doubler	60
Mortgage Loans	61
Installment Purchase Plan	64
Rule of 78s Loan-Interest Rebate	65

**Programs for a fun break**

Chinese Zodiac	68
Piano	70
Secret Letter	72
Shell Game	74
Backward Writer	77
Highball	78
Remarkable Person	79
Computer Guesses Your Secret Number	80

**Graphics programs**

Draw Sketches	83
Dot Sweeper	85
Stock Price Plotter	85
Sine Wave	87
Annual Sales Bar Graph	87
Time Plot	89
Teacher's Grade-Curve Graph	90
Draw Geometric Shapes	91
People Graph	93
Issues Histogram	94
Horizontal Dot-Plot Graph	95



# Introduction

The TRS-80 Model 100 Portable Computer has created more excitement in the microcomputer industry than any machine to come along in years!

With the arrival of this new instrument comes a need for practical, useful software to make it do all of the things you imagine it will do. What can it do? Just about anything you can imagine. The purpose of this book is to point out 44 different directions your thinking might take as you start to plan new uses for your own Micro Executive Work Station.

You've purchased the hardware. You've installed batteries in it, you've read the owner's manual, you know how to turn it on and enter a simple BASIC program. Now . . . what will you do with it? Here are 44 programs to get you started.

We aim to provide software which is easy to type in via the Model 100 keyboard. These programs are complete and ready to run immediately. They have been thoroughly tested to make sure they work perfectly, with no bugs.

These programs are useful in themselves. But we also hope you will find they make good starting points for further development as you learn more and more about how to program your own computer. Read these programs. Type them into your Model 100. Watch them run. Analyze how the lines in the listings cause the computer to step through a sequence of operations to reach a final goal. You'll see and understand how programs are organized, how they work.

Use these practical and fun programs as they are now and then, later, modify them to do more or different work. Expand them to suit your needs as your interests grow.

These programs are designed to be typed directly into your computer, via its keyboard. No other programming is needed to put these software listings to immediate use.

We assume you have read the owner's manual and other instructional materials which came with your Model 100 and with any accessories you may have. You know how to turn on the computer, how to get into the BASIC program-writing mode, how to attach any accessories you may have. If you don't know these beginning steps, please refer to the TRS-80 Model 100 owner's manual.

*You do not have to be a programmer to use these pieces of software. Just type them in, as you find them here, and run them. They will work!*

## **Computer printouts**

To make sure no bugs or errors appear in this book, we have written and tested each of these programs on our own TRS-80 Model 100. We have printed each one on a TRS-80 Line Printer. The hardcopy printout from that line printer is reproduced directly in this book.

The Model 100 operated the printer and listed these programs. No human hands came between the computer and these listings so no retyping errors or proofreading errors have been introduced. You should find that these programs will run exactly as reproduced here.

If, after typing in a program, you get an error message during a run, compare your typed program carefully with the program lines in this book. Remember that every space, punctuation mark, letter, number and symbol counts. Each must be in the computer exactly as in this book.

If you do get an error message, the most likely cause will be found to have been a typing error made while transferring a program from this book to the computer. However, should you find an error in this book, please call it to the attention of the author by sending a postcard or letter to him in care of *ARCsoft Publishers*, P.O. Box 132, Woodsboro, MD 21798 USA. The author will appreciate being able to make any necessary corrections to future editions of this book.

## **Home, school and office**

This book has been organized into three sections for your convenience in locating programs. The first section includes practical programs for use in the home, in the classroom and in the business or professional office or laboratory. Within the first section, the programs have been further subdivided into three categories, including programs of a business interest, personal finance programs, and programs for self-help and self-improvement.

The second section of the book holds what we call programs for a fun break. These are exciting and challenging game programs designed to brighten your day and give you a break from the more serious nature of other computer work.

The third section holds a number of programs designed to display graphics on the computer's LCD screen. These are both practical and informative.

Naturally, these sections, as we have divided the book, are not rigid and exclusive. Undoubtedly, you will find something in one section which will be applicable in a different setting. A business program may be educational. A personal finance program may have business use. Business or finance or personal programs are in the graphics section.

Try them all. They are great fun to run. And they are especially designed to be short so you won't have to spend hours typing one program into your Model 100.

### **Endless running**

Many of the programs in this book will continue to run until you command them off manually, via the BREAK key. You may stop any run, at any time, by pressing the BREAK key on the right side of the keyboard.

### **REMARKs**

As you read through these programs you will notice very few REM, or remarks, statements. The author's training in writing BASIC-language computer programs for battery-powered portable computers included an emphasis on brevity and saving memory space. A sharp editing pencil was in order — and still is!

REMARKs and explanations in software are out. Honing, fine-tuning and waste trimming are in. Use of coding form worksheets for program writing is encouraged. Such worksheets can be found in the publication, *Pocket Computer Program Writing Workbook*, by this author, ISBN 0-86668-817-X, published by *ARCsoft Publishers*. Your objective always should be to make the most efficient use of available memory.

And here's another important note to remember. Even though they may be headed toward the same goal, no two programmers will write the exact same list of BASIC instructions, or program lines, from scratch. As you load these programs into your Model 100, one at a time, you can make modifications to suit your personal needs and interests if you want to. For instance, the exact wording of PRINT statements can be changed. Or, two or more programs can be combined into one grand scheme. Your applications may vary.

By the way, if you want to load more than one of these programs into your Model 100 at the same time, be sure to use different sets of line numbers for different programs.

There cannot be two lines with the same number. For example, only one line can be numbered 10. There can't be two lines numbered 10 in the computer at the same time.

Computer folks use the words ENTER and RETURN interchangeably. The two words usually refer to the keyboard key which signifies to the computer that the end of a line has been reached. In this book, you will use the ENTER key which is located on the right-hand side of the keyboard. For example, you ENTER a program line into the computer by typing it on the keyboard and pressing the ENTER key at the end of the line. Pressing ENTER causes the program line to be placed in the computer's program memory.

### **Other computers**

These programs will run exactly as they are found in this book on any computer using the same Microsoft version of the BASIC language as that used in the TRS-80 Model 100. For example, they will run on the NEC Portable Computers in the PC-8200 series.

Other computers, with various versions of Microsoft BASIC, may require very slight changes to program lines.

Computers with more radically different versions of BASIC may require more extensive changes to program lines.

These are, however, very simple and straightforward BASIC programs which will require little modification to run on any computer set up to be programmed in the BASIC language.

The most significant changes, when needed for other computers, will be in the area of screen graphics. Graphics commands differ from model to model in the microcomputer world. Also, use of multiple-statement lines, the colon, line numbering, spacing, logical tests, multiplication symbols, print statements, and other instructions may differ in other brand-names of computer hardware.

Refer to the owner's manual which came with your non-Model 100 computer to compare its version of the BASIC language with that found in the TRS-80 Model 100.

The author would like to have your suggestions for future editions of this work, or for other titles in this series for the Model 100 Portable Computer line from Radio Shack. The author may be addressed in care of *ARCsoft Publishers*.



## **Standalone vs. subroutine**

The programs in this book may be used as portions of larger lists of instructions to your Model 100 Portable Computer. They can be written in as GOTO or GOSUB objects. When doing so, make appropriate changes to the first line (usually numbered 10 in this book) and to the last line of each program. Remember that no two lines can have the same line number.

In creating a subroutine, remember that every GOSUB must have a RETURN. RETURN must be the last line of every subroutine.

When working one of these programs into a larger set of instructions, be especially careful of your memory or variable names or labels. They must agree with, and fit into, those you are using in the entire program.

## **Learning programming**

These programs are written to be typed into your Model 100 Portable Computer just as you find them here — with no extra programming needed.

Many of the programs, and most of the programming advice, in this book will be of interest to old-hands, as well as newcomers, since we are presenting new twists and more powerful exercises aimed at making your computer do more work, more quickly.

Amidst the 44 programs in this book, you will find countless ideas for using your portable computer. Each piece of software is intended to make you a more-versatile programmer and make your programming chores lighter.

This is not a replacement for the owner's manual which came with your portable computer. You need to read that book thoroughly first, then use this book as a supplement.

Use this book to stimulate your thinking about how to approach various software problems and projects. Use it to get good ideas for new and different approaches to all of your programming goals. While you grow and develop as a program author, modify these programs to make your own computer do even more.

Good programming!

# Making Change

A penny saved is a penny earned. Every businessman is aware of pennies, nickels, dimes, even quarters lost by sales people who can't make correct change. If you have sales people accepting cash away from your register, or if you're too small to have a cash register, use this program to make *correct* change.

This highly unusual program will be exciting for you to type into your computer, and run, even if you aren't a business person.

Key in the amount of the sale and the amount of money tendered by the customer and this software will tell you exactly how many quarters, dimes, nickels and pennies to hand back to the customer.

## Program Listing

```
10 CLS: CLEAR
20 INPUT "NEW SALE AMOUNT IN CENTS "; X
30 INPUT "AMOUNT TENDERED IN CENTS "; T
40 PRINT
90 A=T-X
100 IF A<25 THEN 140
110 Q=Q+1
120 A=A-25
130 GOTO 100
140 IF A<10 THEN 180
150 D=D+1
160 A=A-10
170 GOTO 140
180 IF A<5 THEN 220
190 N=N+1
200 A=A-5
210 GOTO 180
220 P=A
230 PRINT "CORRECT CHANGE IS:"
240 PRINT Q;" QUARTERS"
250 PRINT D;" DICES"
260 PRINT N;" NICKLES"
270 PRINT P;" PENNIES"
280 PRINT "PRESS M FOR MORE OR Q TO QUIT"
```

```
290 K$=INKEY$
300 IF K$="" THEN 290
310 IF K$="M" THEN 10
320 IF K$="Q" THEN END
330 GOTO 280
```

### Sample Run

```
NEW SALE AMOUNT IN CENTS ? 12
NEW SALE AMOUNT IN CENTS ? 12
AMOUNT TENDERED IN CENTS ? 100
CORRECT CHANGE IS:
  3  QUARTERS
  1  DICES
  0  NICKLES
  3  PENNIES
PRESS M FOR MORE OR Q TO QUIT
```

```
NEW SALE AMOUNT IN CENTS ? 77
NEW SALE AMOUNT IN CENTS ? 77
AMOUNT TENDERED IN CENTS ? 100
CORRECT CHANGE IS:
  0  QUARTERS
  2  DICES
  0  NICKLES
  3  PENNIES
PRESS M FOR MORE OR Q TO QUIT
```

```
NEW SALE AMOUNT IN CENTS ? 1
NEW SALE AMOUNT IN CENTS ? 1
AMOUNT TENDERED IN CENTS ? 25
CORRECT CHANGE IS:
  0  QUARTERS
  2  DICES
  0  NICKLES
  4  PENNIES
PRESS M FOR MORE OR Q TO QUIT
```

# Advertising Costs

## Cost Per Thousand

## Cost Per Unit Sold

It's a lot of trouble to determine exactly how much audience you are buying with your advertising dollar. Why not let your computer do all the work?

Suppose your local radio station time salesperson told you he could deliver 51,000 listeners for each \$133 ad you run on his station. And your local newspaper space salesman said he could deliver 160,000 readers for each \$330 ad you run in his newspaper. Which would be the better quantity buy for you?

This program gives you the answers in black and white. The newspaper would cost you about \$2.06 for each 1000 readers while the radio station would cost almost \$2.61 per thousand listeners. Now all you need to decide is which audience you prefer.

By the way, such cost-per-thousand comparisons apply to magazines, TV or any medium as well as radio and newspapers.

This program also computes advertising cost per quantity of units sold. Suppose your favorite newspaper did have the lowest cost-per-thousand so you placed an ad. It cost you \$330. Lots of customers came by to check out your merchandise and you actually sold 77 pieces. What did it cost you to sell each item?

With this program you'll know it cost you \$4.28 in advertising money to sell each unit.

This program runs until you press the BREAK key. To stop the run, press BREAK.

### Program Listing

```
10 CLS: CLEAR
20 GOSUB 900
30 PRINT:PRINT @ 125, "COST PER THOUSAND"
   ;SPACE$(4); "PRESS T"
40 PRINT @ 165, "COST PER UNIT SOLD"
   ;SPACE$(3); "PRESS S"
50 K$=INKEY$
```



```

60 IF K$="" THEN 50
70 IF K$="T" THEN 200
80 IF K$="S" THEN 500
90 PRINT @ 250,"PRESS ONLY T OR S"
100 GOTO 50
200 CLS:GOSUB 900
210 PRINT @ 125,"COST OF AD";SPACE$(4);
:INPUT A
220 PRINT @ 165,"CIRCULATION";SPACE$(3);
:INPUT C
230 M=1000*(A/C)
240 M=INT(100*M+0.5)/100
250 PRINT @ 205,"COST/THOUSAND $";M
260 PRINT @ 245,"PRESS M FOR MORE"
270 IF INKEY$="" THEN 270
280 GOTO 10
500 CLS:GOSUB 900
510 PRINT @ 125,"COST OF AD";SPACE$(8);
:INPUT A
520 PRINT @ 165,"NUMBER UNITS SOLD"
:SPACE$(1);:INPUT U
530 C=A/U
540 C=INT(100*C+0.5)/100
550 PRINT @ 205,"AD COST/UNIT SOLD $";C
560 GOTO 260
890 END
900 LINE (78,2)-(154,20),1,BF
910 LINE(82,6)-(150,16),2,BF
920 PRINT @ 54,"ADVERTISING"
930 RETURN

```

### Sample Run

ADVERTISING

COST PER THOUSAND	PRESS T
COST PER UNIT SOLD	PRESS S

ADVERTISING

COST OF AD	? 133
CIRCULATION	? 51000

COST/THOUSAND \$ 2.61  
PRESS M FOR MORE

### ADVERTISING

COST PER THOUSAND      PRESS T  
COST PER UNIT SOLD      PRESS S

PRESS ONLY T OR S

### ADVERTISING

COST OF AD              ? 2500  
NUMBER UNITS SOLD ? 77  
AD COST/UNIT SOLD \$ 32.47  
PRESS M FOR MORE

## Daily Codes

Businesses everywhere are concerned these days with security. Banks, credit card managers, warehousemen, shipping clerks, office managers, all need private daily codes for internal use to prevent unauthorized admission to storage areas, financial records, private files.

Now you can use your computer to generate a set of secret codes, one for each day of the week. This program generates a series of pseudorandom numbers and displays a table of those numbers alongside names of the days of the week. You can change the list every day or as often as you like because every time you run this program a new set of numbers is generated.

If you don't like a set of numbers, merely press any key on the computer keyboard and the machine will generate and display a new set of daily numbers. As written, this program generates a five-digit number for each day. The random-number generator subroutine starts at line 520 and ends at RETURN.

The program runs until you press the BREAK key to stop it.

By the way, this program also shows how to format the

LCD screen display with a box drawn around information and a two-column display of numbers.

### Program Listing

```
10 CLS: CLEAR
20 PRINT @ 43, "DAILY CODES FOR WEEK OF "
   ;DATE$
30 LINE(16,21)-(210,57),1,B
90 GOSUB 500
100 PRINT @ 124, "MONDAY   ";X:GOSUB 500
110 PRINT @ 140, "TUESDAY  ";X:GOSUB 500
120 PRINT @ 164, "WEDNESDAY";X:GOSUB 500
130 PRINT @ 180, "THURSDAY";X:GOSUB 500
140 PRINT @ 204, "FRIDAY   ";X:GOSUB 500
150 PRINT @ 220, "SATURDAY";X:GOSUB 500
160 PRINT @ 253, "SUNDAY";X
200 IF INKEY$="" THEN 200
210 GOTO 10
400 END
500 FOR Q=1 TO VAL(RIGHT$(TIME$,2))
   :X=INT(100000*RND(1)):NEXT Q
510 IF X<10000 THEN 500
520 RETURN
```

### Sample Run

DAILY CODES FOR WEEK OF 04/10/83

MONDAY	13314	TUESDAY	45574
WEDNESDAY	69184	THURSDAY	72080
FRIDAY	57265	SATURDAY	27207
SUNDAY	73727		

DAILY CODES FOR WEEK OF 04/10/83

MONDAY	57756	TUESDAY	94954
WEDNESDAY	49796	THURSDAY	77256
FRIDAY	45035	SATURDAY	77203
SUNDAY	48860		

# Inventory

This program seems simple at first glance but it allows you to call into play a tremendous amount of the program power and memory capacity built into your Model 100 Portable Computer. With this software, you can keep an inventory of items you have in stock. The inventory includes item name, quantity on hand and a brief description of the item.

This workaholic program uses many of the special features found in your computer, including the ability to do *reverse* printing on the LCD display screen, generating sound, storing in RAMfiles built into your computer memory. The program features multiple erase warnings, automatic return to the computer's main menu when finished, menus and other refinements.

As written here, the program causes the computer to sound a BEEP when the computer wants you to type in some data. You can disable this beep by eliminating the lines which have the BEEP command. If you wish to keep the lines in the program for future use, but disable the BEEP for now, insert a REM or an apostrophe(') between the line number and the BEEP command.

A very powerful feature of the portable computer is the ability it gives you to store data onboard, in RAMfiles, without needing external tape recorders or disk drives. The RAMfiles are portions of the computer's memory set aside to hold such data. The data will be held as long as the memory backup battery is in good working condition. This program stores in, recalls from and adds to RAMfiles.

Adequate warning is given the user before any RAMfiles are erased, as you will see when you type in and run this program. By the way, make sure the CAPS LOCK key is depressed to its *on* position when running this program.

The program is menu driven. That means that, at every step of the way through a run, the programs asks you what you want to do next. It does that by presenting a *menu* of choices available to you at that point. The menus can be seen below in the SAMPLE RUN.

Please note, as you run this program, the length of some data items (number of letters, numbers, symbols, spaces) is limited. Please limit the item name to 8 letters, numbers, symbols and spaces. Limit the quantity to 6 digits. Hold the description to 18 or fewer characters.

## Program Listing

```
10 CLS: CLEAR: RV$=CHR$(27)+"p"  
   : NV$=CHR$(27)+"q"  
100 PRINT @ 55, RV$; "INVENTORY"; NV$  
110 LINE(87,7)-(143,15),1,E  
120 PRINT @ 123, "ADD TO INVENTORY LIST"  
   : SPACE$(5); "PRESS A"  
130 PRINT @ 163, "READ INVENTORY LIST"  
   : SPACE$(7); "PRESS R"  
140 PRINT @ 203, "ERASE INVENTORY LIST"  
   : SPACE$(6); "PRESS E"  
150 PRINT @ 243, "QUIT"; SPACE$(22)  
   : "PRESS Q"  
  
160 BEEP  
200 K$=INKEY$  
210 IF K$="" THEN 200  
220 BEEP  
230 IF K$="A" THEN 400  
240 IF K$="R" THEN 600  
250 IF K$="E" THEN 800  
260 IF K$="Q" THEN 1100  
270 LINE(83,7)-(161,15),2,E  
280 PRINT @ 50, "PRESS ONLY A, R OR Q"  
290 GOTO 200  
400 CLS  
410 PRINT @ 50, RV$; "ADD TO INVENTORY LIST"  
   : NV$  
420 LINE(59,7)-(185,15),1,E  
430 PRINT @ 120, "ITEM NAME:"; : INPUT IT$  
440 IF LEN(IT$)>8 THEN PRINT "PLEASE LIMIT  
   ITEM NAME TO 8 LETTERS": PRINT @ 132  
   , SPACE$(19): GOTO 430  
442 IF LEN(IT$)<8 THEN IT$=IT$+SPACE$(  
   (8-LEN(IT$)))  
445 BEEP  
447 PRINT @ 160, SPACE$(39)  
450 PRINT @ 160, "QUANTITY OF ITEM:";  
   : INPUT QY$  
452 IF LEN(QY$)>6 THEN PRINT "PLEASE LIMIT  
   QUANTITY TO 6 DIGITS": PRINT @ 179  
   , SPACE$(20): GOTO 450
```

```

453 IF LEN(QY$)<6 THEN QY$=QY$+SPACE$
    (6-LEN(QY$))
455 BEEP
458 PRINT @ 200,SPACE$(40)
460 PRINT @ 200,"DESCRIPTION:";:INPUT DS$
462 IF LEN(DS$)>18 THEN PRINT"PLEASE
    LIMIT DESCRIPTION TO 18 LETTERS"
    :PRINT @ 214,SPACE$(26):GOTO 460
463 IF LEN(DS$)<18 THEN DS$=DS$+SPACE$
    (18-LEN(DS$))
465 BEEP
470 RF$=IT$+SPACE$(4)+QY$+SPACE$(4)+DS$
500 OPEN "RAM:INVTRY.DO" FOR APPEND AS 1
510 PRINT#1,RF$
520 CLOSE#1
530 GOTO 10
600 CLS
610 PRINT @ 12,RV$;"READING THE LIST";NV$
620 LINE(71,0)-(71,7),1
630 LINE -(167,7),1
640 LINE -(167,0),1
700 ON ERROR GOTO 795
705 OPEN "RAM:INVTRY.DO" FOR INPUT AS 1
710 PRINT @ 80,"ITEM";SPACE$(8);"QUANTITY"
    ;SPACE$(2);"DESCRIPTION"
715 PRINT @ 120,"-----";SPACE$(8);"-----"
    ;SPACE$(2);"-----"
720 INPUT#1,RF$
730 PRINT @ 160,RF$
750 PRINT @ 240,"PRESS ANY KEY TO SEE
    MORE OF THE LIST"
760 IF INKEY$="" THEN 760
765 BEEP
770 PRINT @ 160,SPACE$(39)
775 IF EOF(1) THEN 785
780 GOTO 710
785 CLOSE#1:CLS
790 PRINT @ 120,"THERE IS NOTHING MORE
    ON THE LIST":FOR T=1 TO 1000:NEXT T
    :GOTO 10
795 PRINT @ 120,"THERE ARE NO ITEMS ON

```



```

THE LIST":FOR T=1 TO 1500:NEXT T:GOTO 10
800 CLS
810 FOR L=1 TO 3
815 LINE(95,7)-(137,15),1,B
820 PRINT @ 56,RV$;"WARNING"
825 FOR T=1 TO 150:NEXT T
830 IF L<3 THEN CLS
835 FOR T=1 TO 25:NEXT T
840 BEEP
845 NEXT L
850 PRINT NV$
860 PRINT @ 125,"YOU WILL ERASE THE
      WHOLE LIST"
870 PRINT @ 205,"TO ERASE THE LIST"
      ;SPACE$(5);"PRESS E"
880 PRINT @ 245,"TO PREVENT ERASING"
      ;SPACE$(4);"PRESS P"
900 K$=INKEY$
910 IF K$="" THEN 900
915 BEEP
920 IF K$="E" THEN 1000
930 IF K$="P" THEN 10
940 GOTO 900
1000 CLS
1010 LINE(95,7)-(137,15),1,B
1020 PRINT @ 56,RV$;"WARNING"
1030 PRINT NV$
1040 PRINT @ 120,"ARE YOU SURE YOU
      WANT TO ERASE"
1050 INPUT"ANSWER YES OR NO ";YN$
1055 BEEP
1060 ON ERROR GOTO 1090
1065 IF YN$="YES" THEN KILL "INVTRY.DO"
      :GOTO 10
1070 IF YN$="NO" THEN 10
1080 GOTO 1050
1090 CLS:PRINT @ 120,"THERE WAS NOTHING
      ON THE LIST TO ERASE":FOR T=1 TO
      1000:NEXT T:GOTO 10
1100 CLS
1110 PRINT @ 55,"END OF RUN"

```

1120 FOR T=1 TO 100:NEXT T  
1130 MENU

### Sample Run

#### INVENTORY

ADD TO INVENTORY LIST	PRESS A
READ INVENTORY LIST	PRESS R
ERASE INVENTORY LIST	PRESS E
QUIT	PRESS Q

#### ADD TO INVENTORY LIST

ITEM NAME:? WIDGET  
QUANTITY OF ITEM:? 12  
DESCRIPTION:? BLACK BOXES

#### READING THE LIST

ITEM	QUANTITY	DESCRIPTION
WIDGET	12	BLACK BOXES

PRESS ANY KEY TO SEE MORE OF THE LIST

#### INVENTORY

ADD TO INVENTORY LIST	PRESS A
READ INVENTORY LIST	PRESS R
ERASE INVENTORY LIST	PRESS E
QUIT	PRESS Q

#### WARNING

YOU WILL ERASE THE WHOLE LIST  
TO ERASE THE LIST PRESS E  
TO PREVENT ERASING PRESS P

#### WARNING

ARE YOU SURE YOU WANT TO ERASE  
ANSWER YES OR NO ? YES

# Daily Receipts Adder

This program allows a businessman to quickly add up his day's receipts, from both wholesale and retail sales as desired.

The computer first will collect wholesale dollars from you. To bypass this or end a list of wholesale-dollar entries, press the ENTER key without any data. By entering "nothing" in this manner, the computer will be forced to move on to the next part of the program. It will go to retail and ask for those dollars from you.

When you finish entering retail dollars, press ENTER with no data and the machine will leave that entry loop. Then it displays a summary of results, including total wholesale dollars, total number of wholesale entries, total retail dollars, total number of retail entries, grand total of all dollars and grand total of all entries.

This program will rerun after results are displayed. Merely press any key on the computer keyboard to make the program start over. The program runs until you press the BREAK key to end.

## Program Listing

```
10 CLS: CLEAR
20 PRINT @ 49, "DAILY RECEIPTS ADDER"
30 PRINT @ 89, STRING$(20, 42)
40 PRINT @ 169, "WHOLESALE $";: INPUT WH$
50 IF WH$="" THEN PRINT @ 169, SPACE$(31)
   :GOTO 100
60 WT=WT+1
70 WD=WD+VAL(WH$)
80 WH$=""
85 PRINT @ 180, SPACE$(19)
90 GOTO 40
100 PRINT @ 169, "RETAIL $";: INPUT RL$
110 IF RL$="" THEN 160
120 RT=RT+1
130 RD=RD+VAL(RL$)
140 RL$=""
145 PRINT @ 177, SPACE$(22)
150 GOTO 100
```

```

160 CLS
170 PRINT WT;"WHOLESALE ENTRIES $";WD
180 PRINT RT;"RETAIL ENTRIES $";RD
195 PRINT STRING$(28,45)
190 PRINT WT+RT;"TOTAL ENTRIES $"
    ;WD+RD
200 PRINT:PRINTSPACE$(1);"PRESS M TO DO
    MORE"
210 IF INKEY$="" THEN 210
220 GOTO 10

```

### Sample Run

```

DAILY RECEIPTS ADDER
*****
WHOLESALE $? 123

```

```

DAILY RECEIPTS ADDER
*****
WHOLESALE $? 398

```

```

DAILY RECEIPTS ADDER
*****
WHOLESALE $?

```

```

DAILY RECEIPTS ADDER
*****
RETAIL $? 682

```

```

DAILY RECEIPTS ADDER
*****
RETAIL $?

```

```

2 WHOLESALE ENTRIES $ 521
1 RETAIL ENTRIES $ 682

```

---

```

3 TOTAL ENTRIES $ 1203

```

```

PRESS M TO DO MORE

```

```

DAILY RECEIPTS ADDER
*****
WHOLESALE $?

```

# **Income Property Cash Flow Analysis**

Here's a handy program to put your computer to work on your real estate investments. Potential gross income, effective gross income, net operating income, and gross spendable cash are the before-tax cash flows of interest.

In computing these cash flows, one first finds the potential gross income by multiplying rent per unit you own times the number of units you own and that times the number of rental payment periods per year. The result will be the rental income you could expect if the property were fully occupied.

Next, you must deduct an allowance for vacancies and rental losses which you will have. This usually is a percentage and the result will be your rent collections. These rent collections would be the same as your effective gross income if you have no other income.

If you have other income, such as receipts from concessions like laundry equipment and other money produced by sources other than rental office space, add it in. The result will be effective gross income.

Now deduct your operating expenses which are those expenditures you must make to keep the property capable of producing the gross income. The result will be the net operating income.

Finally, deduct your annual debt service (interest) on the mortgage and you will find the gross spendable cash.

With this program, you won't have to do all of the math work yourself. Rather, the computer will ask you a few well-chosen questions and it will produce the answers you want immediately.

For an example, see the SAMPLE RUN below. It represents money you receive from an apartment building holding 60 apartments. Each apartment rents for \$250 per month. Some 5 percent of the apartments are vacant on average. The owner pays \$76,855 a year in operating costs. He has a \$700,000 mortgage fully amortized in equal monthly payments for 20 years at 11.5 percent.

Feeding these numbers into the computer, the results are \$180,000 potential gross income; \$9000 vacancy loss; \$171,000 effective gross income; \$94,145 net operating

income; \$89,580.09 annual debt service; and \$4564.91 gross spendable cash.

Upon completion of a run and presentation of its results, the computer will await a press of any key on the keyboard to do another set. It will continue to run until you press the BREAK key to stop it.

### Program Listing

```
10 CLS: CLEAR
20 INPUT "NO. UNITS FOR RENT"; U
30 INPUT "MONTHLY RENT PER UNIT $"; R
40 G=12*R*U
50 INPUT "VACANCY RATE %"; V
60 L=G*(.01*V)
70 INPUT "OTHER INCOME $"; OI
80 EG=G+OI-L
90 INPUT "ANNUAL OPERATING COST $"; OC
100 NI=EG-OC
110 INPUT "MORTGAGE ANNUAL DEBT SERVICE $"; DS
120 SC=NI-DS
130 CLS
140 PRINT "POTENTIAL GROSS INCOME $"; G
150 PRINT "VACANCY LOSS $"; L
160 PRINT "OTHER INCOME $"; OI
170 PRINT "EFFECTIVE GROSS INCOME $"; EG
180 PRINT "NET OPERATING INCOME $"; NI
190 PRINT "MORTGAGE DEBT SERVICE $"; DS
200 PRINT "GROSS SPENDABLE CASH $"; SC
300 IF INKEY$="" THEN 300
310 GOTO 10
```

### Sample Run

```
NO. UNITS FOR RENT? 60
MONTHLY RENT PER UNIT $? 250
VACANCY RATE %? 5
OTHER INCOME $? 0
ANNUAL OPERATING COST $? 76855
MORTGAGE ANNUAL DEBT SERVICE $? 89580.09
```



POTENTIAL GROSS INCOME	\$ 180000
VACANCY LOSS	\$ 9000
OTHER INCOME	\$ 0
EFFECTIVE GROSS INCOME	\$ 171000
NET OPERATING INCOME	\$ 94145
MORTGAGE DEBT SERVICE	\$ 89580.09
GROSS SPENDABLE CASH	\$ 4564.91

## Executive Decision Maker

Ask the computer any question requiring a yes/no answer and you'll get one of 10 possible replies from the machine.

This short, easy-to-type program will give you new answers every time you press any key on the keyboard. The program will continue to run until you stop it by pressing the BREAK key.

### Program Listing

```

10 CLS: CLEAR: RESTORE
20 FOR L=1 TO VAL(RIGHT$(TIME$,1))+1
   : READ D$: NEXT L
30 PRINT @ 119+(((40-LEN(D$))/2),D$
40 IF INKEY$="" THEN 40 ELSE 10
50 DATA YES,FIRE SOMEONE,PASS THE BUCK,
   MAYBE,REORGANIZE,NO,SEE YOUR ANALYST,
   SIT ON IT,FORGET IT,NEVER MIND

```

## Temperature Converter

Convert Celsius to Fahrenheit or Kelvin. Change Fahrenheit to Celsius or Kelvin. See Kelvin temperatures in Celsius or Fahrenheit degrees. This handy program makes the necessary conversions.

At the start of a run, the program presents a menu from which you select the direction of the temperature conversion you need. The computer expects you will press either the F, C or K keys. No other key presses will be accepted. In fact, if you press any other key, the computer will tell you to

limit your commands to only the F, C or K keys.

When you have made your menu selection, the computer will ask for the number of degrees in the information you have. It then will display that temperature in all three systems, Fahrenheit, Celsius and Kelvin.

After the conversion is complete, press any key to do another conversion. The program will run until you press the BREAK key to stop it.

### Program Listing

```
10 CLS: CLEAR
20 PRINT @ 49, "TEMPERATURE CONVERTER"
30 LINE(52,6)-(180,16),1,B
40 PRINT @ 121, "PLEASE SELECT DIRECTION
   OF CONVERSION:"
50 PRINT @ 168, "FROM FAHRENHEIT, PRESS F"
60 PRINT @ 208, "FROM CELSIUS,      PRESS C"
70 PRINT @ 248, "FROM KELVIN,      PRESS K"
80 KY$=INKEY$
90 IF KY$="" THEN 80
100 IF KY$="F" THEN 200
110 IF KY$="C" THEN 300
120 IF KY$="K" THEN 400
130 CLS
140 PRINT @ 126, ">>> PRESS ONLY F, C OR
   K <<<"
150 GOTO 50
200 FOR Z=121 TO 279:PRINT CHR$(127);
   :NEXT Z
210 PRINT @ 128, "FAHRENHEIT TEMPERATURE ";
   :INPUT F
220 C=(F-32)/1.8
230 K=(F+459.67)/1.8
240 GOTO 500
300 FOR Z=121 TO 279:PRINT CHR$(127);
   :NEXT Z
310 PRINT @ 129, "CELSIUS TEMPERATURE ";
   :INPUT C
320 F=1.8*C+32
330 K=C+273.15
340 GOTO 500
```

```

400 FOR Z=121 TO 279:PRINT CHR$(127);
    :NEXT Z
410 PRINT @ 130,"KELVIN TEMPERATURE ";
    :INPUT K
420 F=1.8*K-459.67
430 C=K-273.15
500 CLS
510 PRINT @ 49,"CONVERTED TEMPERATURE"
520 LINE(52,6)-(180,16),1,B
530 PRINT @ 130,"FAHRENHEIT:";F;CHR$(166)
540 PRINT @ 170,"CELSIUS:    ";C;CHR$(166)
550 PRINT @ 210,"KELVIN:     ";K;CHR$(166)
600 IF INKEY$="" THEN 600
610 GOTO 10

```

### Sample Run

#### TEMPERATURE CONVERTER

PLEASE SELECT DIRECTION OF CONVERSION:

FROM FAHRENHEIT, PRESS F  
 FROM CELSIUS, PRESS C  
 FROM KELVIN, PRESS K

FAHRENHEIT TEMPERATURE ? 32

CONVERTED TEMPERATURE

FAHRENHEIT: 32 °  
 CELSIUS: 0 °  
 KELVIN: 273.15 °

#### TEMPERATURE CONVERTER

PLEASE SELECT DIRECTION OF CONVERSION:

FROM FAHRENHEIT, PRESS F  
 FROM CELSIUS, PRESS C  
 FROM KELVIN, PRESS K

>>> PRESS ONLY F, C OR K <<<

FROM FAHRENHEIT, PRESS F  
 FROM CELSIUS, PRESS C  
 FROM KELVIN, PRESS K

CELSIUS TEMPERATURE ? 0

### CONVERTED TEMPERATURE

FAHRENHEIT: 32 °  
CELSIUS: 0 °  
KELVIN: 273.15 °

### TEMPERATURE CONVERTER

PLEASE SELECT DIRECTION OF CONVERSION:

FROM FAHRENHEIT, PRESS F  
FROM CELSIUS, PRESS C  
FROM KELVIN, PRESS K

KELVIN TEMPERATURE ? 273.15

### CONVERTED TEMPERATURE

FAHRENHEIT: 32 °  
CELSIUS: 0 °  
KELVIN: 273.15 °

### TEMPERATURE CONVERTER

PLEASE SELECT DIRECTION OF CONVERSION:

FROM FAHRENHEIT, PRESS F  
FROM CELSIUS, PRESS C  
FROM KELVIN, PRESS K

## Jogger Logger

What a handy way to keep a record of your runs! And with the portable computer, you can take your machine right into the field with you for immediate comparisons to previous jogging outings.

The program allows you to add new run records, review old records, erase all records. When you have completed a day's run, simply turn on the computer, select this program, and press A for ADD NEW RUN RECORD at the first menu. The computer will ask you for the run date. The should reply in the 8-character MM/DD/YY format including the slash-bar (/) symbol.

Next the computer will ask for the distance you ran. You should answer in up to three digits to a maximum of 999

miles. When the computer asks for the length of your run time, reply in minutes.

For example, in our SAMPLE RUN below you'll see that on December 31, 1985 we ran 4 miles in 24 minutes.

If you wish to review your old records, select that option by pressing the R key at the program's main menu. This will result in a display of your old records, one at a time. Press any key on the keyboard to call up the next record.

If you want to erase your entire log, select that option by pressing E at the main menu. The computer won't let you make a mistake easily since it repeatedly queries you to make sure you really do want to erase all of your records.

When you are through using the program, select Q to quit, at the main menu. The run will stop.

This program, as others in this book, calls forth many of the extra-powerful features of the Model 100 Portable Computer. For instance, you will use reverse printing on the LCD display panel. This is accomplished by establishing the reverse-on switch in RV\$ in line 10 and the normal-on switch in NV\$ in line 10. Printing RV\$ before other letters causes them to be in reverse, as in line 100. Following the reverse printing by printing NV\$ causes the display to be switched back to normal printing. For example, in line 100, we switch on reverse printing, then print the words JOGGER LOGGER, then switch off the reverse printing so the display is back to normal.

The program prompts you to type in data by causing the computer to BEEP when information is needed. You can disable this BEEP by simply eliminating the entire line which has a BEEP command. Or, if you wish to keep the line while disabling the BEEP, add a REM or an apostrophe (') between the line number and the BEEP command to turn the line into a REMark statement. REMs are ignored by the computer during a run so the BEEP will be disabled.

Make sure the CAPS LOCK key is depressed to its *on* position when running the program.

The program uses the powerful RAMfiles built into your computer's memory, instead of external tapes or disks. As long as the memory backup battery is working properly, your log will be held in RAMfile memory. You have the power, when running the program, to write to, read from, or erase those RAMfiles. The program gives adequate warning when you are in danger of erasing the files.

When entering the date of a new run, the trap in line 440 prevents you from entering more or less than 8 characters.

### Program Listing

```
100 CLS: CLEAR: RV$=CHR$(27)+"p"
    :NV$=CHR$(27)+"q"
110 PRINT @ 54, RV$; "JOGGER LOGGER"; NV$
120 LINE(83,7)-(161,15),1,B
130 PRINT @ 125, "ADD NEW RUN RECORD"
    :SPACE$(5); "PRESS A"
140 PRINT @ 165, "REVIEW OLD RECORDS"
    :SPACE$(5); "PRESS R"
150 PRINT @ 205, "ERASE ALL RECORDS"
    :SPACE$(6); "PRESS E"
160 PRINT @ 245, "QUIT"; SPACE$(19)
    : "PRESS Q"
170 BEEP
180 K$=INKEY$
190 IF K$="" THEN 200
200 BEEP
210 IF K$="A" THEN 400
220 IF K$="R" THEN 600
230 IF K$="E" THEN 800
240 IF K$="Q" THEN 1100
250 LINE(83,7)-(161,15),2,B
260 PRINT @ 50, "PRESS ONLY A, R OR Q"
270 GOTO 200
280 CLS
290 PRINT @ 51, RV$; "ADD NEW RUN RECORD"
    :NV$
300 LINE(65,7)-(173,15),1,B
310 PRINT @ 120, "RUN DATE (MM/DD/YY)";
    :INPUT DT$
320 IF LEN(DT$)<>8 THEN PRINT "PLEASE
    USE MM/DD/YY FORMAT":PRINT @ 141
    :SPACE$(19):GOTO 310
330 BEEP
340 PRINT @ 160, "DISTANCE RUN (UP TO
    999 MILES)":INPUT DS$
350 BEEP
360 PRINT @ 200, "RUN TIME (IN MINUTES)";
    :INPUT TM$
```



```

465 BEEP
470 RF$=DT$+SPACE$(4)+DS$+SPACE$(8)+TM$
500 OPEN "RAM:RUNS.DO" FOR APPEND AS 1
510 PRINT#1,RF$
520 CLOSE#1
530 GOTO 10
600 CLS
610 PRINT @ 51,RV$;"REVIEW OLD RECORDS"
    ;NV$
620 LINE(65,7)-(173,15),1,B
700 ON ERROR GOTO 795
705 OPEN "RAM:RUNS.DO" FOR INPUT AS 1
710 PRINT @ 120,"DATE";SPACE$(8);"MILES"
    ;SPACE$(4);"MINUTES"
720 INPUT#1,RF$
730 PRINT @ 160,RF$
750 PRINT @ 240,"PRESS ANY KEY TO
    CONTINUE"
760 IF INKEY$="" THEN 760
765 BEEP
770 PRINT @ 160,SPACE$(39)
775 IF EOF(1) THEN 785
780 GOTO 710
785 CLOSE#1:CLS
790 PRINT @ 120,"THERE ARE NO MORE
    RECORDS":FOR T=1 TO 1000:NEXT T
    :GOTO 10
795 PRINT @ 120,"THERE ARE NO OLD
    RECORDS":FOR T=1 TO 1500:NEXT T
    :GOTO 10
800 CLS
810 FOR L=1 TO 4
815 LINE(95,7)-(137,15),1,B
820 PRINT @ 56,RV$;"WARNING"
825 FOR T=1 TO 250:NEXT T
830 IF L<4 THEN CLS
835 FOR T=1 TO 50:NEXT T
840 BEEP
845 NEXT L
850 PRINT NV$
860 PRINT @ 123,"YOU WILL ERASE ALL OF
    YOUR RECORDS"

```

```

870 PRINT @ 205,"TO ERASE ALL RECORDS
    PRESS E"
880 PRINT @ 245,"TO PREVENT ERASING
    PRESS P"
900 K$=INKEY$
910 IF K$="" THEN 900
915 BEEP
920 IF K$="E" THEN 1000
930 IF K$="P" THEN 10
940 GOTO 900
1000 CLS
1010 LINE(95,7)-(137,15),1,B
1020 PRINT @ 56,RV$;"WARNING"
1030 PRINT NV$
1040 PRINT @ 120,"ARE YOU SURE YOU WANT
    TO ERASE"
1050 INPUT"YES OR NO ";YN$
1055 BEEP
1060 ON ERROR GOTO 1090
1065 IF YN$="YES" THEN KILL "RUNS.DO"
    :GOTO 10
1070 IF YN$="NO" THEN 10
1080 GOTO 1050
1090 CLS:PRINT @ 120,"THERE WERE NO
    RECORDS TO ERASE":FOR T=1 TO 1000
    :NEXT T:GOTO 10
1100 CLS
1110 PRINT @ 55,"END OF RUN"

```

## Sample Run

### JOGGER LOGGER

ADD NEW RUN RECORD	PRESS A
REVIEW OLD RECORDS	PRESS R
ERASE ALL RECORDS	PRESS E
QUIT	PRESS Q

PRESS ONLY A, R OR Q

ADD NEW RUN RECORD	PRESS A
REVIEW OLD RECORDS	PRESS R
ERASE ALL RECORDS	PRESS E
QUIT	PRESS Q

ADD NEW RUN RECORD

RUN DATE (MM/DD/YY)? 12/31/85  
DISTANCE RUN (UP TO 999 MILES)? 4  
RUN TIME (IN MINUTES)? 24

REVIEW OLD RECORDS

DATE	MILES	MINUTES
12/31/85	4	24

PRESS ANY KEY TO CONTINUE

JOGGER LOGGER

ADD NEW RUN RECORD	PRESS A
REVIEW OLD RECORDS	PRESS R
ERASE ALL RECORDS	PRESS E
QUIT	PRESS Q

WARNING

YOU WILL ERASE ALL OF YOUR RECORDS

TO ERASE ALL RECORDS	PRESS E
TO PREVENT ERASING	PRESS P

WARNING

ARE YOU SURE YOU WANT TO ERASE  
YES OR NO ? YES

JOGGER LOGGER

ADD NEW RUN RECORD	PRESS A
REVIEW OLD RECORDS	PRESS R
ERASE ALL RECORDS	PRESS E
QUIT	PRESS Q

END OF RUN

Ok

# Dual-Time Universal Clock

Here's an exciting clock program that's good anywhere in the USA at any time of the year because it takes time zone and standard times into account. It presents both local time and Universal time, on the LCD display panel, at the same time. And it runs endlessly until you press the BREAK key. Very convenient!

When you first start a program run, the computer will display a menu asking for information on your time zone. You press E for eastern, C for central, M for mountain or P for pacific.

Next, the computer asks whether you are on standard time or daylight savings time. Press S for standard or D for daylight savings.

At that point the clock will start running, displaying your local time with proper label on the left and UTC universal time on the right.

## Program Listing

```
10 CLS: CLEAR: BB=1
20 GOSUB 900
30 PRINT @ 126, "WHAT IS YOUR U.S. TIME
   ZONE?"
40 PRINT @ 201, "EASTERN,  PRESS E"
50 PRINT @ 223, "CENTRAL,  PRESS C"
60 PRINT @ 241, "MOUNTAIN, PRESS M"
70 PRINT @ 263, "PACIFIC,  PRESS P"
80 TZ$=INKEY$
90 IF TZ$="" THEN 80
100 IF TZ$="E" OR TZ$="C" OR TZ$="M"
    OR TZ$="P" THEN 120
110 GOTO 80
120 CLS
130 GOSUB 900
140 PRINT @ 122, "ON STANDARD OR DAYLIGHT
   SAVINGS TIME?"
150 PRINT @ 211, "STANDARD, PRESS S"
160 PRINT @ 251, "DAYLIGHT, PRESS D"
170 DL$=INKEY$
180 IF DL$="" THEN 170
190 IF DL$="S" OR DL$="D" THEN 210
```

```

200 GOTO 170
210 CLS:BB=2
220 GOSUB 900
230 GOSUB 800
240 PRINT @ 170,TZ$+DL$+"T"
250 PRINT @ 187,"UTC"
270 GOSUB 700
280 NT$=STR$(HR)+RIGHT$(TIME$,6)
290 IF LEN(STR$(HR))<3 THEN NT$="0"+
    RIGHT$(STR$(HR),1)+RIGHT$(TIME$,6)
300 PRINT @ 207,TIME$
310 PRINT @ 225,NT$
320 GOTO 270
690 END
700 HR=VAL(LEFT$(TIME$,2))+C
710 IF HR>23 THEN HR=HR-24
720 RETURN
800 IF DL$="S" THEN C=1
810 IF TZ$="E" THEN C=C+4
820 IF TZ$="C" THEN C=C+5
830 IF TZ$="M" THEN C=C+6
840 IF TZ$="P" THEN C=C+7
850 RETURN
900 LINE(74,4)-(164,18),1,B
910 LINE(75,5)-(163,17),1,B
920 IF BB=1 THEN PRINT @ 53,"TIME
    CONVERTER"
930 IF BB=2 THEN PRINT @ 53,"UNIVERSAL
    TIME"
940 RETURN

```

### Sample Run

```

                TIME CONVERTER
        WHAT IS YOUR U.S. TIME ZONE?

EASTERN,  PRESS E          CENTRAL,  PRESS C
MOUNTAIN, PRESS M          PACIFIC,  PRESS P

                TIME CONVERTER
        ON STANDARD OR DAYLIGHT SAVINGS TIME?

                STANDARD, PRESS S
                DAYLIGHT, PRESS D

```

UNIVERSAL TIME

EDT  
07:58:20

UTC  
11:58:20

## Bubble Sort

This program sorts names, labels, words, letters into alphabetical order, A to Z. As written here it will accept up to 26 items, sort them, and print a list of those names horizontally across the screen in alphabetical order. Then it prints the list in original entry order for comparison. It shows that the items really were sorted and none were lost!

You may type in a list of up to 26 items. If you type in fewer than 26, press ENTER without data to exit the entry loop and get on with sorting.

After results are displayed, press any key to do a new set. The program runs continuously until you press the BREAK key to stop it.

If you wish to change the number of names or items accepted, change the number 26 in lines 20, 30, 200, 230, and the number 25 in line 100. If you wish to remove the printing of the original order of entry from the program, delete lines 55, 230, 240, and 250.

### Program Listing

```
10 CLS: CLEAR
20 DIM M$(26): DIM D$(26)
30 FOR L=1 TO 26
40 INPUT "NAME"; M$(L)
50 IF M$(L)=" " THEN 70
55 D$(L)=M$(L)
60 NEXT L
70 CLS
80 PRINT "SORTING NOW"
90 T=0
100 FOR L=1 TO 25
110 IF M$(L)<=M$(L+1) THEN 160
120 E$=M$(L)
130 M$(L)=M$(L+1)
140 M$(L+1)=E$
```



```

150 T=1
160 NEXT L
170 IF T=1 THEN 90
180 CLS
190 BEEP
200 FOR L=1 TO 26
210 IF M$(L)<>" THEN PRINT M$(L); " ";
220 NEXT L
225 PRINT
230 FOR L=1 TO 26
240 IF D$(L)<>" THEN PRINT D$(L); " ";
250 NEXT L
300 IF INKEY$="" THEN 300
310 GOTO 10

```

### Sample Run

```

NAME? SMITH
NAME? JONES
NAME? ANDREWS
NAME? LEWIS
NAME? BROWN
NAME?

SORTING NOW

ANDREWS BROWN JONES LEWIS SMITH
SMITH JONES ANDREWS LEWIS BROWN

```

## Day Of The Year

This program reports the number of the day of the year for any day you enter.

See the menus in the SAMPLE RUN below. The program asks for the date, giving you the choice of today's date by pressing T or some other date by pressing X.

If you select today's date, the program moves immediately to computation and presentation of results. If you press X, it asks for that date and wants to receive it in the MM/DD/YYYY format.

When a result is printed on the LCD display panel, press any key to do another. The program runs continuously until you stop it by pressing the BREAK key.

## Program Listing

```
10 CLEAR
20 GOSUB 500
40 PRINT @ 134,"SELECT ONE:"
50 PRINT @ 168,"TODAY'S DATE (PRESS T)"
60 PRINT @ 208,"OTHER DATE (PRESS X)"
70 KY$=INKEY$
80 IF KY$="" THEN 70
90 IF KY$="T" THEN 200
100 IF KY$="X" THEN 400
110 PRINT @ 250,"PRESS ONLY T OR X"
120 GOTO 40
200 D$=DATE$
210 J=VAL(LEFT$(D$,2))
220 K=VAL(MID$(D$,4,2))
230 IF KY$="T" THEN I=VAL("19"+(RIGHT$(D$,2)))
235 IF KY$="X" THEN I=VAL(RIGHT$(D$,4))
240 N=INT(3055*(J+2)/100)-91
250 L=0
260 IF I=4*INT(I/4) THEN L=1
270 IF I=100*INT(I/100) THEN L=0
280 IF I=400*INT(I/400) THEN L=1
290 IF J>2 THEN N=N-2+L
300 N=N+K
310 GOTO 600
400 GOSUB 500
410 PRINT @ 126,"TYPE IN DATE AS
MM/DD/YYYY"
420 PRINT @ 172," ";
430 INPUT D$
440 GOTO 210
500 CLS
510 PRINT @ 51,CHR$(27)+"p";"DAY OF THE
YEAR";CHR$(27)+"q"
520 LINE(65,7)-(155,15),1,B
530 RETURN
600 GOSUB 500
610 PRINT @ 129,"DAY";K;"OF MONTH";J
620 PRINT @ 161,"IS DAY NUMBER";N
;"IN THE YEAR";I
```

```
630 PRINT @ 245,"PRESS ANY KEY TO DO  
  ANOTHER"  
640 IF INKEY$="" THEN 640  
650 GOTO 10
```

### **Sample Run**

```
      DAY OF THE YEAR  
      SELECT ONE:  
      TODAY'S DATE (PRESS T)  
      OTHER DATE   (PRESS X)  
  
      DAY OF THE YEAR  
      DAY 11 OF MONTH 8  
IS DAY NUMBER 224 IN THE YEAR 1984  
  
      PRESS ANY KEY TO DO ANOTHER  
  
      DAY OF THE YEAR  
      SELECT ONE:  
      TODAY'S DATE (PRESS T)  
      OTHER DATE   (PRESS X)  
  
      DAY OF THE YEAR  
      TYPE IN DATE AS MM/DD/YYYY  
      ? 12/31/1984  
  
      DAY OF THE YEAR  
      DAY 31 OF MONTH 12  
IS DAY NUMBER 366 IN THE YEAR 1984  
  
      PRESS ANY KEY TO DO ANOTHER
```

## **Name 'N Note List**

Here's a handy way to keep track of friends' phone numbers or employees' payroll numbers or family charge-account numbers. In fact, this list keeper will be useful anywhere you have a list of pairs of data to be stored together.

The program allows you to type in names and attach notes to those names. The names, with their attached notes, may be typed in, in any random order. The program auto-

matically arranges the names, with their attached notes, in alphabetical order.

The program gives you the option of printing the list on the computer's LCD display screen or on paper via a printer. Naturally, you must have an appropriate line printer or daisy-wheel printer attached to the computer to use the print-on-paper option.

The notes could be telephone numbers, code numbers, your word comments, addresses, membership status, employee status, or anything else you might care to append to someone's name on a list. Hold the length of the name down to 19 or fewer characters including letters, numbers, symbols and spaces. Similarly, hold the length of the note to 19 or fewer characters.

As you run this program, note the interesting use of graphics to highlight the questions, then the sorting message, and the various messages from the computer as it displays results of its work.

The program uses a bubble-sort routine to arrange the names in alphabetical order. The more names in the list, the longer time it will take to sort. As written here, the program accepts a maximum of 50 names, controlled by the number 50 in lines 15, 20, 400 and 660 and the number 49 in line 200. You may change these numbers to alter the capacity for names. The program alone uses about 1200 bytes of memory.

### Program Listing

```
10 CLS: CLEAR
15 DIM M$(50): DIM D$(50): DIM Q$(50)
20 FOR L=1 TO 50
25 PRINT @ 43, "TYPE NAME"; L;
30 LINE(16,6)-(223,16),1,B
35 INPUT Q$(L)
50 IF Q$(L)="" THEN 90
55 LINE(16,6)-(223,16),2,B
60 PRINT @ 83, "TYPE NOTE"; L;
65 LINE(16,6)-(223,24),1,B
70 INPUT D$(L)
75 M$(L)=Q$(L)+" "+D$(L)
80 CLS
85 NEXT L
```

```

90 CLS
95 LINE(75,22)-(145,32),1,B
100 LINE(74,21)-(146,33),1,B
105 PRINT @ 133,"SORTING NOW"
110 T=0
200 FOR L=1 TO 49
210 IF M$(L)<=M$(L+1) THEN 260
220 E$=M$(L)
230 M$(L)=M$(L+1)
240 M$(L+1)=E$
250 T=1
260 NEXT L
270 IF T=1 THEN 110
280 CLS
290 BEEP
300 CLS:PRINT
310 PRINT"WHICH WAY DO YOU WANT THE
    LIST PRINTED?"
315 PRINT
320 PRINT"ON THE LCD DISPLAY PANEL
    (PRESS L)"
330 PRINT"ON PAPER BY THE PRINTER
    (PRESS P)"
340 K$=INKEY$
350 IF K$="" THEN 340
360 IF K$="L" THEN CLS:GOTO 400
370 IF K$="P" THEN 600
380 PRINT"PRESS ONLY L OR P"
390 GOTO 340
400 FOR L=1 TO 50
410 IF M$(L)<>"" THEN PRINT M$(L):P=P+1
420 IF P<6 THEN 460
430 PRINT"    >>>PRESS ANY KEY TO READ
    MORE<<<"
440 IF INKEY$="" THEN 440
450 P=0
460 NEXT L
470 P=0
500 PRINT "LIST END:PRESS M FOR MORE OR
    Q TO QUIT"
510 K$=INKEY$
520 IF K$="" THEN 510

```

```

530 IF K$="M" THEN 300
540 IF K$="Q" THEN END
550 GOTO 500
600 CLS
610 PRINT @ 88,"PLEASE CONNECT A PRINTER"
620 LINE(46,14)-(192,24),1,B
630 PRINT @ 171,"PRESS D WHEN DONE"
640 KZ$=INKEY$
650 IF KZ$<>"D" THEN 640
660 FOR L=1 TO 50
670 IF M$(L)<>" " THEN LPRINT M$(L)
680 NEXT L
690 CLS:PRINT:GOTO 500

```

### Sample Run

```

TYPE NAME 1 ?
TYPE NAME 1 ? EDWARD
TYPE NOTE 1 ?
TYPE NAME 1 ? EDWARD
TYPE NOTE 1 ? (800) 555-1234

TYPE NAME 2 ?
TYPE NAME 2 ? JANE
TYPE NAME 2 ? JANE
TYPE NOTE 2 ? (900) 555-8765
TYPE NAME 3 ?

          SORTING NOW

WHICH WAY DO YOU WANT THE LIST PRINTED?
ON THE LCD DISPLAY PANEL (PRESS L)
ON PAPER BY THE PRINTER (PRESS P)

WHICH WAY DO YOU WANT THE LIST PRINTED?
ON THE LCD DISPLAY PANEL (PRESS L)
ON PAPER BY THE PRINTER (PRESS P)
PRESS ONLY L OR P

EDWARD (800) 555-1234
JANE (900) 555-8765
LIST END:PRESS M FOR MORE OR Q TO QUIT

```



WHICH WAY DO YOU WANT THE LIST PRINTED?  
ON THE LCD DISPLAY PANEL (PRESS L)  
ON PAPER BY THE PRINTER (PRESS P)

PLEASE CONNECT A PRINTER

PRESS D WHEN DONE

EDWARD (800) 555-1234

JANE (900) 555-8765

LIST END:PRESS M FOR MORE OR Q TO QUIT

## Code Breaker

This fancy encoding and decoding program allows you to write secret messages in code and decode secret messages written in the same code. The use of a powerful *code key* format means only the largest computers in the world would have a chance at cracking your code and then only with great difficulty. For all intents and purposes, the code generated by this program is unbreakable.

When you start a run of this program, the computer will ask whether you wish to encode or decode a message. Press E to encode a new message or D to decode an existing message.

Next, type in a secret number as a *code key* which only you, and others you tell, will know. This number could be one previously specified or obtained by counting the number of words on a page or the number of pickets in a fence or some other fixed number of items. The person using this program to decipher your message must have access to the same *code key* number, one way or another.

Any *code key* number is okay, however a larger number will take longer for the computer to assimilate.

After you have typed in the *code key* number, the computer will ask you to type in a one-line message. For use on our small screen, we have suggested the message be only one line in length. Actually, it can be up to 255 characters long. The computer will display the message, letter by letter, as you type it in.

When the message has been completely entered, press ENTER and the encoded (or decoded) message will be

displayed instantly on the screen. If you have encoded a new message, copy it off on paper and send it to a friend. If he has the same software in his computer and knowledge of the *code key* he will be able to type it into his computer and see it instantly decoded.

The same sequence of events happens while decoding as during encoding. Just type in the message to be decoded, press ENTER, and you get an instant translation.

The letter transpositions used in encoding rely on random numbers generated by the computer so the *code key* which controls the random-number generator is vital to being able to decipher someone else's message. It would be extremely difficult, if not impossible, to unravel the letter transpositions without the *code key*.

After you see and read a translated message, the display will hold until you press any key on the keyboard. Pressing any key then will make it go back to the start of the program where you will have the option of encoding or decoding a new message. The program runs endlessly until you press the BREAK key.

Imagine taking this powerful encoding and decoding ability with you, anywhere in the world in your portable computer!

### Program Listing

```
100 CLS: CLEAR
110 DIM M$(255)
120 GOSUB 700
130 PRINT @ 122, "WILL YOU ENCODE OR
    DECODE A MESSAGE?"
140 PRINT @ 211, "TO ENCODE, PRESS E"
150 PRINT @ 251, "TO DECODE, PRESS D"
160 KY$=INKEY$
170 IF KY$="" THEN 160
180 IF KY$="E" OR KY$="D" THEN 200
190 GOTO 160
200 CLS
210 GOSUB 700
220 IF KY$="E" THEN F$="EN"
230 IF KY$="D" THEN F$="DE"
240 PRINT @ 125, " ";
250 PRINT "WHAT IS THE "; F$; " CODING KEY";
```

```

260 INPUT S
270 FOR L=1 TO S:N=INT(60*RND(1)):NEXT L
280 CLS
300 GOSUB 700
310 PRINT @ 127,"TYPE IN A ONE LINE
    MESSAGE"
320 PRINT
325 Q=1
330 K$=""
340 K#=INKEY$
350 IF K$="" THEN 340
360 IF ASC(K$)=13 THEN 410
370 PRINT K$;
380 M$(Q)=STR$(ASC(K$))
390 Q=Q+1
400 GOTO 340
410 IF KY$="E" THEN 600
500 PRINT @ 239," ";
505 FOR L=1 TO Q-1
510 Z=VAL(M$(L))-INT(60*RND(1))
520 IF Z<32 THEN Z=Z+59
530 PRINT CHR$(Z);
540 NEXT L
550 IF INKEY$="" THEN 550
560 GOTO 100
600 PRINT @ 239," ";
605 FOR L=1 TO Q-1
610 Z=VAL(M$(L))+INT(60*RND(1))
620 IF Z>90 THEN Z=Z-59
630 PRINT CHR$(Z);
640 NEXT L
650 GOTO 550
700 LINE (80,4)-(158,18),1,B
710 LINE (81,5)-(157,17),1,B
720 PRINT @ 54,"CODE BREAKER"
730 PRINT
740 RETURN

```

### Sample Run

CODE BREAKER

WILL YOU ENCODE OR DECODE A MESSAGE?

TO ENCODE, PRESS E  
TO DECODE, PRESS D

CODE BREAKER

WHAT IS THE ENCODING KEY? 37

CODE BREAKER

TYPE IN A ONE LINE MESSAGE  
THE QUICK BROWN FOX JUMPED OVER THE HIGH

CODE BREAKER

TYPE IN A ONE LINE MESSAGE  
THE QUICK BROWN FOX JUMPED OVER THE HIG  
!II'\*J/0%,B4S:9XQGFLWDC S-//B-( GNB/GMN?

CODE BREAKER

WILL YOU ENCODE OR DECODE A MESSAGE?  
TO ENCODE, PRESS E  
TO DECODE, PRESS D

CODE BREAKER

WHAT IS THE DECODING KEY? 37

CODE BREAKER

TYPE IN A ONE LINE MESSAGE  
!II'\*J/0%,B4S:9XQGFLWDC S-//B-( GNB/GMN?

CODE BREAKER

TYPE IN A ONE LINE MESSAGE  
!II'\*J/0%,B4S:9XQGFLWDC S-//B-( GNB/GMN  
THE QUICK BROWN FOX JUMPED OVER THE HIGH

## Decimal-To-Binary Number Converter

Type in any decimal number and see it immediately converted to binary format. After conversion, press any key to do a new conversion. The program runs endlessly until you press BREAK.

### Program Listing

```
10 CLEAR
```

```

20 GOSUB 500
30 PRINT
40 INPUT"DECIMAL NUMBER":N
50 IF N<0 OR N<>INT(N) THEN PRINT"NUMBER
   MUST BE A POSITIVE INTEGER":GOTO 30
60 X=N:N$=""
70 Q=INT(X/2):R=X-2*Q
80 N$=MID$("01",R+1,1)+N$
90 IF Q<>0 THEN X=Q:GOTO 70
300 PRINT"BINARY NUMBER   ":N$
310 PRINT @ 240,"PRESS ANY KEY TO DO
   ANOTHER"
320 IF INKEY$="" THEN 320
330 GOTO 10
490 END
500 CLS
510 PRINT @ 43,CHR$(27)+"p";"DECIMAL TO
   BINARY NUMBER CONVERSION":CHR$(27)+"q"
520 LINE(18,7)-(227,15),1,B
530 RETURN

```

### Sample Run

DECIMAL TO BINARY NUMBER CONVERSION

```

DECIMAL NUMBER? 14
BINARY NUMBER   1110

```

PRESS ANY KEY TO DO ANOTHER

## Hour Clock

Here's an odd new kind of clock which tells you the hour at a glance. The program uses graphics to simulate a clock face on the computer screen, including the hour numbers of 1 to 12. The current hour of the day is indicated by flashing the appropriate hour number.

Upon the start of a run, the program will cause the computer to display the clock face and flash the hour continuously until you press the BREAK key. The computer

constantly updates the hour display by checking TIME\$ at program line 400.

## Program Listing

```
10 CLS: CLEAR
100 DATA 62,104,146,184,222
110 DATA 260,218,176,134,96,58,20
200 FOR L=1 TO 12
210 READ P
220 PRINT @ P, CHR$(239)
230 FOR T=1 TO 100:NEXT T
240 PRINT @ P,L
250 NEXT L
300 RESTORE
400 HR=VAL(LEFT$(TIME$,2))
410 FOR L=1 TO HR
420 READ P
430 NEXT L
440 PRINT @ P, CHR$(239)
450 RESTORE
460 FOR T=1 TO 100:NEXT T
470 PRINT @ P," "
480 FOR T=1 TO 100:NEXT T
500 GOTO 400
```

## Code Practice Oscillator

To learn Morse code you had to buy an oscillator and a key and get a friend who knew the code to sit down for a few hours to send dots and dashes to you. No more!

Now you have a friend in your computer. It will send letters and numbers all day and all night, if you like, without break. In fact, this program runs continuously until you stop it with the BREAK key.

CAI means *computer-assisted instruction* and this program is CAI at its finest. The program makes the computer generate the dots and dashes of the International Morse code so you can hear, see and learn that code. The computer will show very large-size dots and dashes on the screen and sound loud dits and dahs.

After displaying and sound off, the computer will ask what the character it showed was. It awaits your answer which you give by pressing the appropriate key on the keyboard. If your answer is wrong, the computer will say so. If it is right, it will say so. Either way, it will say what the correct answer is.

The speed and tone are fixed so you don't have to worry about anything except learning to recognize the letters as they are sent to you by the computer. This is great for the beginner who needs to learn the International Morse code. After right or wrong, press any key to do another.

### Program Listing

```
10 CLS
20 FOR Q=1 TO VAL(RIGHT$(TIME$,2))
   :RN=INT(37*RND(1)):NEXT Q
30 IF RN<1 THEN 20
40 FOR L=1 TO (11*RN)
50 READ C
60 NEXT L
70 PRINT @ 90, " ";
100 FOR L=1 TO 11
110 READ C
120 IF C=2 THEN C$=CHR$(239)+CHR$(239)
   +CHR$(239):SOUND 2500,24
130 IF C=1 THEN C$=CHR$(239):SOUND 2500,8
140 IF C=0 THEN C$=CHR$(224):SOUND 2500,0
150 PRINT C$;
160 NEXT L
200 PRINT @ 171,"WHAT WAS THAT CHARACTER?"
210 A$=INKEY$
220 IF A$="" THEN 210
230 IF A$=CHR$(C) THEN PRINT @ 171,"YES, "
   ;CHR$(C); " IS CORRECT";SPACE$(7) ELSE
   PRINT @ 171,"NO, IT WAS ";CHR$(C)
   ;SPACE$(12)
240 PRINT
250 PRINT
260 PRINT @ 251,"PRESS ANY KEY FOR
   ANOTHER"

300 IF INKEY$="" THEN 300
```



```

310 RESTORE
320 GOTO 10
500 END
590 DATA 0,0,0,0,0,0,0,0,0,0,0
600 DATA 1,0,2,0,0,0,0,0,0,0,65
610 DATA 2,0,1,0,1,0,1,0,0,0,66
620 DATA 2,0,1,0,2,0,1,0,0,0,67
630 DATA 2,0,1,0,1,0,0,0,0,0,68
640 DATA 1,0,0,0,0,0,0,0,0,0,69
650 DATA 1,0,1,0,2,0,1,0,0,0,70
660 DATA 2,0,2,0,1,0,0,0,0,0,71
670 DATA 1,0,1,0,1,0,1,0,0,0,72
680 DATA 1,0,1,0,0,0,0,0,0,0,73
690 DATA 1,0,2,0,2,0,2,0,0,0,74
700 DATA 2,0,1,0,2,0,0,0,0,0,75
710 DATA 1,0,2,0,1,0,1,0,0,0,76
720 DATA 2,0,2,0,0,0,0,0,0,0,77
730 DATA 2,0,1,0,0,0,0,0,0,0,78
740 DATA 2,0,2,0,2,0,0,0,0,0,79
750 DATA 1,0,2,0,2,0,1,0,0,0,80
760 DATA 2,0,2,0,1,0,2,0,0,0,81
770 DATA 1,0,2,0,1,0,0,0,0,0,82
780 DATA 1,0,1,0,1,0,0,0,0,0,83
790 DATA 2,0,0,0,0,0,0,0,0,0,84
800 DATA 1,0,1,0,2,0,0,0,0,0,85
810 DATA 1,0,1,0,1,0,2,0,0,0,86
820 DATA 1,0,2,0,2,0,0,0,0,0,87
830 DATA 2,0,1,0,1,0,2,0,0,0,88
840 DATA 2,0,1,0,2,0,2,0,0,0,89
850 DATA 2,0,2,0,1,0,1,0,0,0,90
860 DATA 2,0,2,0,2,0,2,0,2,0,48
870 DATA 1,0,2,0,2,0,2,0,2,0,49
880 DATA 1,0,1,0,2,0,2,0,2,0,50
890 DATA 1,0,1,0,1,0,2,0,2,0,51
900 DATA 1,0,1,0,1,0,1,0,2,0,52
910 DATA 1,0,1,0,1,0,1,0,1,0,53
920 DATA 2,0,1,0,1,0,1,0,1,0,54
930 DATA 2,0,2,0,1,0,1,0,1,0,55
940 DATA 2,0,2,0,2,0,1,0,1,0,56
950 DATA 2,0,2,0,2,0,2,0,1,0,57

```

## Sample Run

■ ■ ■■■■

WHAT WAS THAT CHARACTER?

■ ■ ■■■■

NO, IT WAS U

PRESS ANY KEY FOR ANOTHER

■ ■■■■ ■

WHAT WAS THAT CHARACTER?

■ ■■■■ ■

NO, IT WAS R

PRESS ANY KEY FOR ANOTHER

■■■■ ■ ■ ■■■■

YES, X IS CORRECT

PRESS ANY KEY FOR ANOTHER

■ ■ ■ ■ ■

YES, 5 IS CORRECT

PRESS ANY KEY FOR ANOTHER

■ ■■■■ ■■■■

YES, W IS CORRECT

PRESS ANY KEY FOR ANOTHER

■■■■ ■■■■ ■■■■ ■ ■

NO, IT WAS 8

PRESS ANY KEY FOR ANOTHER

■■■■ ■ ■

YES, D IS CORRECT

PRESS ANY KEY FOR ANOTHER

## Random Number Tip

Sometimes the "randomness" of the random numbers generated by the computer isn't good enough. Here's how to "reseed" the random number generator to get less-predictable numbers.

Lines 20 and 50 cause the random number always to be printed in the same place on the display. Line 40 causes the "reseeding" and generation of new and different sets of random numbers.

When you run this program, the result will be the display of a random number. Press any key to display a new random number. The program runs until you press BREAK.

### Program Listing

```
10 CLS
20 PRINT @ 130,"RANDOM NUMBER:"
30 CLEAR
40 FOR Z=1 TO VAL(RIGHT$(TIME$,2))
   :P=INT(100*RND(1)):NEXT Z
50 PRINT @ 145,P
60 IF INKEY$="" THEN 60
70 GOTO 30
```

## Alarm Clock

Here's a handy way to turn your computer into an alarm clock! It will await a previously-specified hour and then sound an alert.

You set the alarm time. The computer constantly compares that time with the actual time on its internal clock and beeps when the alarm time is reached.

When you start the program run, the computer will ask for the alarm time. You type it in the HH:MM:SS format. The computer gives you a chance to change your mind by asking if the time you just typed in is okay. Press Y for yes or N for no.

After the alarm sounds, the computer asks if you wish to have the same alarm time tomorrow or set a new alarm time. Press S for the same time tomorrow or C to change the next

alarm time. Note that the computer uses a 24-hour clock so you can set the alarm to sound at the same time every day. The program runs until you press BREAK.

### Program Listing

```
10 GOSUB 500
20 PRINT @ 160,"TYPE ALARM TIME IN THIS
   FORMAT";CHR$(154);" 12:12:12"
30 INPUT AT$
40 GOSUB 500
50 PRINT @ 120,"IS THIS CORRECT? "
   ;CHR$(154);" ";AT$
60 YN$=INPUT$(1)
70 IF YN$="Y" THEN 100
80 IF YN$="N" THEN 10
90 GOTO 60
100 GOSUB 500
110 PRINT @ 133,TIME$
120 IF AT$=TIME$ THEN 200
130 GOTO 110
200 FOR L=1 TO 10
210 FOR P=500 TO 10 STEP -50
220 SOUND P,1
230 NEXT P
240 FOR P=10 TO 500 STEP 50
250 SOUND P,1
260 NEXT P
270 NEXT L
280 GOSUB 500
290 PRINT @ 127,TIME$;" ";DAY$;" ";DATE$
300 PRINT @ 200,"SAME TIME TOMORROW OR
   CHANGE THE TIME?"
340 PRINT"PRESS S (SAME) OR C (CHANGE)"
350 SC$=INKEY$
360 IF SC$="" THEN 350
370 IF SC$="S" THEN 100
380 IF SC$="C" THEN 10
390 GOTO 350
400 END
500 CLS
510 LINE(68,6)-(140,16),1,B
```

```
520 PRINT @ 52,"ALARM CLOCK"
530 RETURN
```

### Sample Run

```
ALARM CLOCK
TYPE ALARM TIME IN THIS FORMAT 12:12:12
? 07:30:00
```

```
ALARM CLOCK

IS THIS CORRECT? 07:30:00
```

```
ALARM CLOCK

08:25:13 Sun 04/10/83
```

```
SAME TIME TOMORROW OR CHANGE THE TIME?
PRESS S (SAME) OR C (CHANGE)
```

## Radio Frequency & Wavelength

Let your computer convert frequency to wavelength and wavelength to frequency. Upon running this program, press F to compute frequency or W to compute wavelength. After conversion press M to do more. The program will run continuously until you press the BREAK key.

### Program Listing

```
10 CLS: CLEAR
20 GOSUB 800
30 PRINT @ 125,"COMPUTE FREQUENCY"
   ;SPACE$(4);"PRESS F"
40 PRINT @ 165,"COMPUTE WAVELENGTH"
   ;SPACE$(3);"PRESS W"
50 K$=INKEY$
60 IF K$="" THEN 50
70 IF K$="W" THEN 200
80 IF K$="F" THEN 500
90 PRINT @ 250,"PRESS ONLY F OR W"
100 GOTO 50
```

```

200 CLS:GOSUB 800
210 PRINT @ 125,"FREQUENCY (MHz)"
    ;SPACE$(4);:INPUT F
220 W=300/F
230 PRINT @ 165,F;"MHz =" ;W;"METERS"
240 PRINT @ 245,"PRESS M FOR MORE"
250 IF INKEY#="" THEN 250
260 GOTO 10
500 CLS:GOSUB 800
510 PRINT @ 125,"WAVELENGTH (METERS)"
    ;SPACE$(1);:INPUT W
520 F=300/W
530 PRINT @ 165,W;"METERS =" ;F;"MHz"
540 GOTO 240
800 LINE(24,2)-(202,20),1,BF
810 LINE(28,6)-(198,16),2,BF
820 PRINT @ 45,"RADIO FREQUENCY &
    WAVELENGTH"
830 RETURN

```

### Program Listing

```

RADIO FREQUENCY & WAVELENGTH
COMPUTE FREQUENCY      PRESS F
COMPUTE WAVELENGTH     PRESS W

```

```

RADIO FREQUENCY & WAVELENGTH
WAVELENGTH (METERS) ? 21
21 METERS = 14.285714285714 MHz

```

PRESS M FOR MORE

```

RADIO FREQUENCY & WAVELENGTH
COMPUTE FREQUENCY      PRESS F
COMPUTE WAVELENGTH     PRESS W

```

PRESS ONLY F OR W

```

RADIO FREQUENCY & WAVELENGTH
FREQUENCY (MHz)       ? 151
151 MHz = 1.9867549668874 METERS

```

PRESS M FOR MORE

# How Money Grows

My, oh my, how your money grows when you put it in a savings account at a certain annual percentage rate of interest, compounded monthly, for a number of months.

This program asks for information on the initial principal saved, annual interest percentage rate, and number of months. The result is a moving display of the changes in principal as the months pass. The computer holds at the final month's display so you can see what you would have at that time.

Lines 40 to 60 take in the data. Lines 70 to 120 do the computation and display results. You can change the length of time of the pause in scrolling by changing the number 250 in the time-delay loop in line 110.

Break the hold at the end of a run and start over by pressing any key. The program will run until you press BREAK to stop it.

## Program Listing

```
10 CLS: CLEAR
20 PRINT @ 53, "MONEY GROWTH"
30 LINE (76,6)-(150,16), 1, B
40 PRINT: INPUT "PRINCIPAL $ " ; P
50 INPUT "ANNUAL INTEREST PERCENT " ; R
60 INPUT "NUMBER OF MONTHS " ; M
65 PRINT
70 FOR Q=1 TO M
80 I=(P*(0.01*R))/12
90 F=P+I
95 Z=INT(P*100+0.5)/100
100 PRINT "AFTER MONTH " ; Q ; " = $" ; Z
110 FOR T=1 TO 250: NEXT T
120 NEXT Q
130 IF INKEY#="" THEN 130
140 GOTO 10
```

## Sample Run

```
                MONEY GROWTH
PRINCIPAL $           ? 1000
ANNUAL INTEREST PERCENT ? 8
NUMBER OF MONTHS      ? 6
```



AFTER MONTH	1	= \$	1006.67
AFTER MONTH	2	= \$	1013.38
AFTER MONTH	3	= \$	1020.13
AFTER MONTH	4	= \$	1026.93
AFTER MONTH	5	= \$	1033.78
AFTER MONTH	6	= \$	1040.67

## Deposit Doubler

There's an old rule of thumb used by some bankers. It holds that money left on deposit in an interest-bearing account will *double* in value as interest compounds.

If you don't take this *Rule of 72nds* too seriously it can be a handy gauge to figure roughly the number of years you need to leave your money in a bank account (or other interest-bearing system) to see it double in value.

The rule holds that the number of years to double is about equal to the number 72 divided by the annual interest rate. The actual conversion is done in line 130 of this program. The rest of the program lines are input and output generators.

### Program Listing

```

10 CLS: CLEAR
20 RV$=CHR$(27)+"P": NV$=CHR$(27)+"Q"
30 PRINT @ 55, RV$; "DEPOSIT DOUBLER"; NV$
40 LINE(90,7)-(179,15), 1, B
50 LINE(88,5)-(181,17), 1, B
100 PRINT @ 120, "WHAT IS ANNUAL INTEREST
    PERCENT";: INPUT I
110 PRINT @ 120, SPACE$(40)
120 PRINT @ 120, "AT "; I; " PERCENT INTEREST"
130 Y=72/I
140 IF Y>INT(Y) THEN 150 ELSE 180
150 D=Y-INT(Y)
160 IF D>0.5 THEN Y=Y+1
170 Y=INT(Y)
180 PRINT @ 160, "MONEY DOUBLES IN ABOUT "
    ; Y; " YEARS"
200 PRINT @ 240, "PRESS "; RV$; "ENTER"; NV$

```

```

; " FOR MORE OR ";RV$;"Q";NV$;" TO QUIT"
205 LINE(35,47)-(65,55),1,B
207 LINE(143,47)-(149,55),1,B
210 K$=INKEY$
220 IF K$="" THEN 210
230 IF K$="Q" OR K$="q" THEN 300
240 IF ASC(K$)=13 THEN 10
250 GOTO 210
300 CLS
310 LINE (77,1)-(155,21),1,BF
320 LINE (81,5)-(151,17),2,BF
330 PRINT @ 54,"TERMINATION"
340 END

```

## Sample Run

```

DEPOSIT DOUBLER
WHAT IS ANNUAL INTEREST PERCENT? 15
DEPOSIT DOUBLER

AT 15 PERCENT INTEREST
MONEY DOUBLES IN ABOUT 5 YEARS

PRESS ENTER FOR MORE OR Q TO QUIT

```

## Mortgage Loans

Planning to buy a new home? Thinking of refinancing the old mortgage? This program computes either mortgage payment amount, number of payments or amount borrowed.

When you first run this program, you are presented with a menu of work options. If you wish to find the amount of payment on a certain mortgage, press P. If you wish to compute the number of payments to complete a mortgage, press N. To recall the original loan amount of an existing mortgage, press V.

If you wish to determine the monthly mortgage payment, the computer will ask for the original loan amount in dollars, the percentage interest rate, and the number of months in the life of the mortgage.

To find number of payments, the computer will require

the original loan amount, the interest rate, and the monthly payment amount. To recreate the original amount of a mortgage, tell the computer payment, interest rate and number of months in the original agreement.

After each computation and presentation of results, the computer is ready to do another. Press any key on the keyboard to start over. Or press BREAK to quit.

### Program Listing

```
10 CLS: CLEAR
20 TT$="MORTGAGE LOANS"
30 GOSUB 800
40 PRINT @ 127, "WHICH DO YOU WISH TO FIND?"
50 PRINT @ 163, "MORTGAGE PAYMENT AMOUNT
  (PRESS P)"
60 PRINT @ 203, "NUMBER OF PAYMENTS
  (PRESS N)"
70 PRINT @ 243, "ORIGINAL LOAN AMOUNT
  (PRESS V)"
80 K$=INKEY$
90 IF K$="" THEN 80
100 IF K$="P" THEN 200
110 IF K$="N" THEN 400
120 IF K$="V" THEN 600
130 CLS: BEEP
140 PRINT @ 89, "PRESS ONLY P, N, OR V"
150 GOTO 50
200 TT$="MORTGAGE PAYMENT"
210 GOSUB 800
220 PRINT
230 INPUT "ORIGINAL LOAN AMOUNT $"; V
240 INPUT "INTEREST RATE PERCENT "; I
250 LET I=0.01*(I/12)
260 INPUT "NUMBER OF MONTHS "; N
270 PRINT
280  $P=V*(I/(1-(1+I)^{-N}))$ 
290  $P=INT(100*P+0.5)/100$ 
300 PRINT "MORTGAGE PAYMENT $"; P; "<<<"
310 PRINT
320 PRINT "FOR MORE PRESS ANY KEY"
330 K$=INKEY$
```

```

340 IF K$="" THEN 330
350 GOTO 10
400 TT$="NUMBER OF PAYMENTS"
410 GOSUB 800
420 PRINT
430 INPUT"ORIGINAL LOAN AMOUNT      $";V
440 INPUT"INTEREST RATE PERCENT    %";I
450 I=0.01*(I/12)
460 INPUT"MORTGAGE PAYMENT AMOUNT  $";P
470 PRINT
480 N=-((LOG(1-I*V/P))/(LOG(1+I)))
500 PRINT"NUMBER OF MONTHS ";N;"<<<"
510 GOTO 310
600 TT$="MORTGAGE LOAN AMOUNT"
610 GOSUB 800
620 PRINT
630 INPUT"MORTGAGE PAYMENT AMOUNT  $";P
640 INPUT"INTEREST RATE PERCENT    %";I
650 I=0.01*(I/12)
660 INPUT"NUMBER OF MONTHS          ";N
670 PRINT
680 V=P*((1-(1+I)^-N)/I)
690 V=INT(100*V+0.5)/100
700 PRINT"LOAN AMOUNT                $"
    ;V;"<<<"
710 GOTO 310
790 END
800 CLS:BEEP
810 TL=LEN(TT$)
820 CT=INT((40-TL)/2)
830 BL=TL*6+4
840 BS=(240-BL)/2
850 LINE(BS,6)-(BS+BL-2,16),1,B
860 PRINT @ 40+CT,TT$
870 RETURN

```

### Sample Run

```

                MORTGAGE LOANS
      WHICH DO YOU WISH TO FIND?
MORTGAGE PAYMENT AMOUNT  (PRESS P)
NUMBER OF PAYMENTS      (PRESS N)
ORIGINAL LOAN AMOUNT     (PRESS V)

```

## MORTGAGE PAYMENT

ORIGINAL LOAN AMOUNT \$? 79500  
INTEREST RATE PERCENT ? 9.75  
NUMBER OF MONTHS ? 360

ORIGINAL LOAN AMOUNT \$? 79500  
INTEREST RATE PERCENT ? 9.75  
NUMBER OF MONTHS ? 360

MORTGAGE PAYMENT \$ 683.03 <<<

FOR MORE PRESS ANY KEY

## Installment Purchase Plan

What can be more confusing than the terms on your credit card? It can be a real hassle trying to get a payment figure from a store or loan company or bank for an installment purchase. With this program, you won't have to call or visit a lender to discover just how much your payment is going to be.

The computer will tell you how much you will have to pay on your installment-credit loan if you give it a few pieces of pertinent information.

Type the program listing below into your computer and RUN it. The computer will ask for the list price of the item you are buying. It will ask for the amount of the down payment you will be making at the time of the purchase. It will ask over how many payments you wish the loan spread. And, finally, it will ask for the annual interest rate percentage of the loan.

After completing its work the computer will tell you the amount of the monthly payment you will have to make on the loan. Then it will wait to do another set of numbers. To do more, press any key on the keyboard. To quit, press the BREAK key.

### Program Listing

```
10 CLS: CLEAR  
20 LINE (26,3)-(201,19),1,BF
```

```

30 LINE (29,6)-(198,16),2,BF
50 PRINT @ 45,"INSTALLMENT PURCHASE
  PAYMENT"
60 PRINT:PRINT"LIST PRICE $";SPACE$(9);
  :INPUT L
70 PRINT"DOWN PAYMENT $";SPACE$(7);
  :INPUT D
80 PRINT"NUMBER OF PAYMENTS";SPACE$(3);
  :INPUT N
90 PRINT"ANNUAL INPUT PERCENT";SPACE$(1);
  :INPUT I
100 I=(0.01*I)/12
110 F=(L-D)*(I/(1-(1/((1+I)^N))))
120 P=INT(100*F+0.5)/100
130 PRINT
200 PRINT "PAYMENT";SPACE$(14);"$";P
210 IF INKEY$="" THEN 210
220 GOTO 10

```

### Sample Run

#### INSTALLMENT PURCHASE PAYMENT

```

LIST PRICE $           ? 10000
DOWN PAYMENT $         ? 2500
NUMBER OF PAYMENTS     ? 48
ANNUAL INPUT PERCENT   ? 12

LIST PRICE $           ? 10000
DOWN PAYMENT $         ? 2500
NUMBER OF PAYMENTS     ? 48
ANNUAL INPUT PERCENT   ? 12

PAYMENT                $ 197.5

```

## Rule of 78's Loan Interest Rebate

Paying off your consumer loan? Want to know if you are due a refund on interest and how much it will be? This pro-

gram will allow the computer to compute the amount of interest you could overpay and the remaining balance of principal you must pay to satisfy the loan.

If you pay off the loan by paying all pre-computed payments, use this program to find how much refund, or *rebate*, is due you. If you haven't paid off the loan yet, use this program to determine the amount of principal to pay off without overpaying interest.

Using what bankers call the *Rule of 78's*, you must supply the number of the current payment, the total number of installment payments for which the loan originally was written, and the original total finance charge or amount of interest dollars.

See the SAMPLE RUN below. As you can see, the computer will ask for the necessary information and then provide the answer you want. When the computation and final presentation of results has been made, the computer stands by, awaiting your order to do another set of numbers. You accomplish this by pressing any key on the computer's keyboard. Or you may press BREAK to end the run.

### Program Listing

```
10 CLS: CLEAR
100 INPUT "ORIGINAL NO. OF MONTHS IN LOAN"
    ;N
105 IF N=<0 THEN 100
110 INPUT "PAYMENT NO. WHEN PREPAYMENT
    OCCURS" ;K
115 IF K>N OR K=<0 THEN 110
120 INPUT "ORIGINAL TOTAL FINANCE CHARGE
    $" ;P
125 IF P=<0 THEN 120
130 INPUT "MONTHLY PAYMENT $" ;PM
135 IF PM=<0 THEN 130
200 I=((2*(N-K+1))/(N*(N+1)))*P
210 RB=((N-K)*I)/2
220 BL=((N-K)*PM)-RB
230 II=INT(I*100+0.5)/100
240 R=INT(RB*100+0.5)/100
250 B=INT(BL*100+0.5)/100
300 CLS
```



```

310 PRINT"RULE OF 78'S LOAN INTEREST
    REBATE"
320 PRINT
330 PRINT"ORIGINAL FINANCE CHARGE $" ;P
340 PRINT"ORIGINAL NO. PAYMENTS    " ;N
350 PRINT"PREPAYMENT MONTH NO.     " ;K
360 PRINT"REBATE DUE                $" ;R
370 PRINT"PRINCIPAL OUTSTANDING    $" ;B
400 IF INKEY$="" THEN 400
410 GOTO 10

```

### Sample Run

```

ORIGINAL NO. OF MONTHS IN LOAN? 48
PAYMENT NO. WHEN PREPAYMENT OCCURS? 24
ORIGINAL TOTAL FINANCE CHARGE $? 1000
MONTHLY PAYMENT $? 750

```

#### RULE OF 78'S LOAN INTEREST REBATE

```

ORIGINAL FINANCE CHARGE $ 1000
ORIGINAL NO. PAYMENTS      48
PREPAYMENT MONTH NO.      24
REBATE DUE                  $ 255.1
PRINCIPAL OUTSTANDING     $ 17744.9

```

```

ORIGINAL NO. OF MONTHS IN LOAN? 24
PAYMENT NO. WHEN PREPAYMENT OCCURS? 18
ORIGINAL TOTAL FINANCE CHARGE $? 100
MONTHLY PAYMENT $? 50

```

#### RULE OF 78'S LOAN INTEREST REBATE

```

ORIGINAL FINANCE CHARGE $ 100
ORIGINAL NO. PAYMENTS    24
PREPAYMENT MONTH NO.    18
REBATE DUE                $ 7
PRINCIPAL OUTSTANDING    $ 293

```

# Chinese Zodiac

The most-famous oriental Zodiac calendar is divided into twelve-year groups. Each year has a different animal sign. The sign under which a person is born is believed to determine the circumstances of his life and the kind of person he is.

The animals are rat, ox, tiger, rabbit, dragon, snake, horse, sheep, monkey, cock, dog and boar.

Run this program. When asked by the computer, type in the year of your birth. The computer will determine which animal sign corresponds to your birth year and present you with information concerning your own circumstances.

The computer names the animal, describes your personality traits, tells which other animal signs are compatible with your own, and tells which animal signs are your opposite, or least compatible.

This fun game runs endlessly. After presenting information, it awaits the press of any keyboard key to start over. Press BREAK to stop.

## Program Listing

```
10 CLEAR:GOSUB 900
20 PRINT:INPUT"IN WHAT YEAR WERE YOU BORN"
   :Y$
30 X$=RIGHT$(Y$,2)
40 N=VAL(X$)
50 IF (N/12)=INT(N/12) THEN 200
60 IF ((N-1)/12)=INT((N-1)/12) THEN 250
70 IF ((N-2)/12)=INT((N-2)/12) THEN 300
80 IF ((N-3)/12)=INT((N-3)/12) THEN 350
90 IF ((N-4)/12)=INT((N-4)/12) THEN 400
100 IF ((N-5)/12)=INT((N-5)/12) THEN 450
110 IF ((N-6)/12)=INT((N-6)/12) THEN 500
120 IF ((N-7)/12)=INT((N-7)/12) THEN 550
130 IF ((N-8)/12)=INT((N-8)/12) THEN 600
140 IF((N-9)/12)=INT((N-9)/12) THEN 650
150 IF ((N-10)/12)=INT((N-10)/12) THEN 700
160 IF ((N-11)/12)=INT((N-11)/12) THEN 750
170 GOTO 10
200 PRINT A$;"RAT"
210 PRINT B$;"AMBITIOUS, SINCERE"
```

```
220 PRINT C$;"DRAGON, MONKEY"
230 PRINT D$;"HORSE"
240 GOTO 800
250 PRINT A$;"OX"
260 PRINT B$;"A LEADER, BRIGHT, CHEERFUL"
270 PRINT C$;"SNAKE, COCK"
280 PRINT D$;"SHEEP"
290 GOTO 800
300 PRINT A$;"TIGER"
310 PRINT B$;"SENSITIVE, COURAGEOUS"
320 PRINT C$;"HORSE, DOG"
330 PRINT D$;"MONKEY"
340 GOTO 800
350 PRINT A$;"RABBIT"
360 PRINT B$;"TALENTED, AFFECTIONATE"
370 PRINT C$;"SHEEP, BOAR"
380 PRINT D$;"COCK"
390 GOTO 800
400 PRINT A$;"DRAGON"
410 PRINT B$;"ROBUST, PASSIONATE"
420 PRINT C$;"MONKEY, RAT"
430 PRINT D$;"DOG"
440 GOTO 800
450 PRINT A$;"SNAKE"
460 PRINT B$;"INTENSE, WITH WISDOM"
470 PRINT C$;"OX, COCK"
480 PRINT D$;"BOAR"
490 GOTO 800
500 PRINT A$;"HORSE"
510 PRINT B$;"ATTRACTIVE, POPULAR"
520 PRINT C$;"TIGER, DOG"
530 PRINT D$;"RAT"
540 GOTO 800
550 PRINT A$;"SHEEP"
560 PRINT B$;"STYLISH, PRIVATE"
570 PRINT C$;"BOAR, RABBIT"
580 PRINT D$;"OX"
590 GOTO 800
600 PRINT A$;"MONKEY"
610 PRINT B$;"PERSUASIVE, INTELLIGENT"
620 PRINT C$;"DRAGON, RAT"
630 PRINT D$;"TIGER"
```

```

640 GOTO 800
650 PRINT A$;"COCK"
660 PRINT B$;"A PIONEER, SEEKING TRUTH"
670 PRINT C$;"SNAKE, OX"
680 PRINT D$;"RABBIT"
690 GOTO 800
700 PRINT A$;"DOG"
710 PRINT B$;"GENEROUS, LOYAL"
720 PRINT C$;"HORSE, TIGER"
730 PRINT D$;"DRAGON"
740 GOTO 800
750 PRINT A$;"BOAR"
760 PRINT B$;"GALANT, NOBLE"
770 PRINT C$;"RABBIT, SHEEP"
780 PRINT D$;"SNAKE"
790 GOTO 800
800 PRINT:PRINT"FOR MORE PRESS ANY KEY"
810 IF INKEY$="" THEN 810
820 GOTO 10
890 END
900 CLS
910 LINE (67,3)-(159,19),1,BF
920 LINE (70,6)-(156,16),2,BF
930 PRINT @ 52,"CHINESE ZODIAC"
940 A$="THE YEAR OF THE "
950 B$="YOU ARE "
960 C$="COMPATIBLE WITH "
970 D$="OPPOSITE OF "
980 RETURN

```

## Piano

Here's how to turn your computer into a simple piano and create some musical fun!

As written, this program causes the computer to play the notes, C,D,E,F,G,A, and B. And the computer displays the letter played on the screen.

The computer responds to your action in pressing either the C,D,E,F,G,A, or B letter keys on the keyboard. Press Q to turn off the piano and end the run.

For instance, if you press the letter C key, at line 130 in the program that action is translated into a number 1174 which is stored in memory P. At line 210, the computer retrieves the number 1174 from memory P and makes the sound it associates with that number. Lines 130 through 190 allow the computer to know whether you have pressed C, D, E, F, G, A, or B and to store the appropriate number in memory P.

You may cause the computer to store other numbers in memory P and, thus, make different sounds as you play the piano. Here's a handy chart of notes and octaves associated with specific numbers.

	Octaves				
	1	2	3	4	5
Notes					
G	12538	6269	3134	1567	783
G#	11836	5918	2959	1479	739
A	11172	5586	2793	1396	698
A#	10544	5272	2636	1318	659
B	9952	4968	2484	1244	622
C	9394	4697	2348	1174	587
C#	8866	4433	2216	1108	554
D	8368	4184	2092	1046	523
D#	7900	3950	1975	987	493
E	7456	3718	1864	932	466
F	7032	3516	1758	879	439
F#	6642	3321	1660	830	415

The duration or length of the note can be changed by changing the number 5 in line 210.

### Program Listing

```

10 CLS: CLEAR
20 R$=CHR$(27)+"P":N$=CHR$(27)+"q"
30 PRINT @ 57,R$;"PIANO";N$
40 LINE(101,7)-(131,15),1,B

```

```

50 LINE(99,5)-(133,17),1,B
60 LINE(97,3)-(135,19),1,B
70 PRINT @ 124,"PRESS KEYS TO PLAY
   THESE NOTES:"
80 PRINT @ 173,"C D E F G A B"
90 LINE (113,47)-(119,55),1,B
100 K$=""
110 K$=INKEY$
120 IF K$="" THEN 110
130 IF K$="C" THEN P=1174:GOTO 210
140 IF K$="D" THEN P=1046:GOTO 210
150 IF K$="E" THEN P=932:GOTO 210
160 IF K$="F" THEN P=879:GOTO 210
170 IF K$="G" THEN P=783:GOTO 210
180 IF K$="A" THEN P=698:GOTO 210
190 IF K$="B" THEN P=622:GOTO 210
195 IF K$="Q" THEN 300
200 GOTO 110
210 SOUND P,5
220 PRINT @ 259,R$;K$;N$
230 GOTO 110
300 PRINT @ 120,SPACE$(40)
310 PRINT @ 160,SPACE$(40)
320 LINE(113,47)-(119,55),2,BF
330 PRINT @ 203,"YOU HAVE JUST TURNED
   OFF THE PIANO"

```

## Secret Letter

Guess-the-number games can be great fun but they're old-hat in the world of personal computers. So we've come up with something that captures that old fun and adds a more challenging, more exciting play. This is the guess-the-secret-letter game.

The computer ponders a moment, then thinks of a letter. It knows that secret letter but it doesn't let you know. Your job is to figure it out.

It starts by displaying the full alphabet of 26 letters on its screen. You make a guess by pressing a letter key. No need to press ENTER.

The computer responds with a CORRECT or WRONG

message. If wrong, it displays the entire alphabet again but this time *without* the letter or letters you have chosen previously. By the way, the tricky part is in having the computer come back and display the entire alphabet without the chosen letter or letters. This is accomplished in the subroutine in lines 700 to 810.

You will note there is a processing delay of 5 or 6 seconds so the computer sends out a wait message.

The computer uses its random-number generator to select the mystery letter. This is done by selecting an ASCII number from 65 to 90, in the subroutine at lines 500 to 520. Number 65 corresponds to the letter A and 90 to letter Z.

In playing the game, you make your selection by pressing a letter key, anywhere from A through Z. No need to press ENTER.

The computer keeps track of your number of attempts and tells you, after each try, how many incorrect guesses you have made. If you make a correct guess, the computer will tell you how many tries it took to get it right.

It provides an interesting afternoon's diversion, consuming many hours of fun. The game plays continuously until you press the BREAK key.

It uses 681 bytes of Model 100 memory. Only a powerful computer with an extensive BASIC vocabulary, such as the Model 100, could do all this!

## Program Listing

```
10 CLS: CLEAR
20 DIM N(26), V(26)
30 GOSUB 500
40 GOTO 160
50 GOSUB 700
60 GOSUB 600
70 PRINT "GUESS THE LETTER"
80 GOSUB 600
90 PRINT: PRINT B$: PRINT
100 LET G$=INKEY$
110 IF G$="" THEN 100
120 CLS: LET T=T+1
130 LET N(T)=ASC(G$)
140 IF ASC(G$)=P THEN 300
150 PRINT: PRINT "WRONG "; T; " TIMES"
```



```

160 PRINT:PRINT"PLEASE STANDBY"
170 PRINT"PROCESSING WILL TAKE 6 SECONDS"
180 GOTO 50
300 PRINT"CORRECT"
310 PRINT CHR$(P); " IS THE LETTER"
320 PRINT
400 PRINT"YOU GOT IT IN ";T;" TRIES"
410 PRINT
420 CLEAR:GOTO 20
430 END
500 FOR W=1 TO VAL(RIGHT$(TIME$,2))
    :P=INT(91*RND(1)):NEXT W
510 IF P<65 THEN 500
520 RETURN
600 FOR Q=1 TO 16
610 PRINT "*";
620 NEXT Q
630 PRINT
640 RETURN
700 LET B$=""
710 FOR Q=1 TO 26
720 IF N(Q)=0 THEN V(Q)=Q+64
730 FOR Z=1 TO 26
740 IF V(Q)=N(Z) THEN V(Q)=0
750 NEXT Z
760 LET B$=B$+CHR$(V(Q))
770 NEXT Q
780 PRINT
790 PRINT"THANK YOU FOR WAITING"
800 PRINT
810 RETURN

```

## Shell Game

Hey, Man! Here's a great new game which can only be played on a powerful computer such as yours.

The dark squares on the display screen are cups. Under one is a letter P, representing a pea. But which cup covers the pea? You guess by pressing the appropriate number key, either 1, 2 or 3. No need to press ENTER. The computer shuffles its cups after each guess.

Upon first starting this program, the computer will display a menu asking you to select an option. Press R to read the rules. Press P to play the game. Press Q to quit or end the game. You can return to the main menu at any time by pressing the ESCape key.

The number of right and wrong guesses is shown on the screen. See how much you can win in 10 minutes of play. Then see how much your friends can win in the same length of time. Or you can compete by number of games played. Or play solitaire by yourself. The object is to locate the pea! It may or may not move after each guess, depending upon the secret decision of the computer. Go for it!

### Program Listing

```
10 CLS: CLEAR
20 RV$=CHR$(27)+"P": NV$=CHR$(27)+"Q"
100 PRINT @ 55, RV$; "SHELL GAME": NV$
110 LINE(89,7)-(149,15), 1, B
120 LINE(87,5)-(151,17), 1, B
130 PRINT @ 130, "TO PLAY"; SPACE$(5)
    ; "PRESS P"
140 PRINT @ 170, "FOR RULES"; SPACE$(3)
    ; "PRESS R"
145 PRINT @ 210, "TO QUIT"; SPACE$(5)
    ; "PRESS Q"
150 K$=INKEY$
160 IF K$="" THEN 150
170 IF K$="P" THEN 200
180 IF K$="R" THEN 1100
185 IF K$="Q" THEN 1300
190 GOTO 150
200 PRINT @ 130, SPACE$(19): PRINT @ 170
    , SPACE$(19): PRINT @ 210, SPACE$(19)
210 PRINT @ 130, "1": PRINT @ 140, "2"
    : PRINT @ 150, "3"
220 GOSUB 900
300 K$=INKEY$
310 IF K$="" THEN 300
315 IF ASC(K$)=27 THEN 10
320 IF VAL(K$)<1 OR VAL(K$)>4 THEN 300
330 GOSUB 1000
```

```

340 GOSUB 500
350 ON X GOSUB 800,700,600
360 FOR T=1 TO 500:NEXT T
400 IF VAL(K$)=X THEN R=R+1:GOTO 420
410 W=W+1
420 PRINT @ 40,R;"RIGHT"
430 PRINT @ 71,W;"WRONG"
440 GOTO 220
500 FOR Q=1 TO VAL(RIGHT$(TIME$,2))
    :X=INT(4*RND(1)):NEXT Q
510 IF X<1 THEN 500
520 RETURN
600 LINE(55,37)-(69,50),1,BF
610 LINE(115,37)-(129,50),1,BF
620 LINE(175,37)-(189,50),1,B
630 PRINT @ 230,"P"
640 RETURN
700 LINE(55,37)-(69,50),1,BF
710 LINE(115,37)-(129,50),1,B
720 LINE(175,37)-(189,50),1,BF
730 PRINT @ 220,"P"
740 RETURN
800 LINE(55,37)-(69,50),1,B
810 LINE(115,37)-(129,50),1,BF
820 LINE(175,37)-(189,50),1,BF
830 PRINT @ 210,"P"
840 RETURN
900 LINE(55,37)-(69,50),1,BF
910 LINE(115,37)-(129,50),1,BF
920 LINE(175,37)-(189,50),1,BF
930 RETURN
1000 LINE(55,37)-(69,50),2,BF
1010 LINE(115,37)-(129,50),2,BF
1020 LINE(175,37)-(189,50),2,BF
1030 RETURN
1100 CLS
1110 PRINT @ 56,RV$;"RULES";NV$
1120 LINE(95,7)-(125,15),1,B
1130 LINE(93,5)-(127,17),1,B
1140 PRINT:PRINT"A 'PEA' IS UNDER ONE
    CUP? WHICH: NUMBER 1, 2 OR 3? THE
    COMPUTER SHUFFLES CUPS"

```

```

1150 PRINT "AFTER EACH GUESS. TO GUESS,
        PRESS NUMBER OF CUP." ; SPACE$(9)
        ; "(NOW PRESS ANY KEY)"
1160 IF INKEY$="" THEN 1160
1170 PRINT @ 120, SPACE$(156)
1180 PRINT @120, "COMPUTER SHOWS NUMBER
        OF RIGHT GUESSES & WRONG GUESSES. ";
1190 PRINT "TO END THE GAME ANYTIME,
        PRESS ESCAPE KEY." ; SPACE$(3) ; "(NOW
        PRESS ANY KEY)"
1200 IF INKEY$="" THEN 1200
1210 CLS:GOTO 100
1300 CLS
1310 PRINT @ 55, RV$ ; "END OF GAME" ; NV$
1320 LINE(89,7)-(155,15),1,B
1330 LINE(87,5)-(157,17),1,B
1340 END

```

## Backward Writer

Type in a message of up to 85 characters in response to the computer's request. The computer will turn your message around and print it backwards on the LCD display panel.

The 85 characters includes letters, numbers, symbols and spaces. You may use any combination. For a test, try it with the 26 letters of the alphabet as we have done in our SAMPLE RUN below. Type Z to A in backwards order. You will be amazed at the speed with which your computer turns it around!

### Program Listing

```

10 CLS: CLEAR
20 INPUT "TYPE A MESSAGE OF UP TO 85
        CHARACTERS " ; A$
30 IF A$="" THEN 20
40 L=LEN(A$)
50 FOR J=L TO 1 STEP -1
60 B$=B$+MID$(A$,J,1)
70 NEXT J

```

```

80 CLS
90 PRINT B$
100 IF INKEY$="" THEN 100
110 GOTO 10

```

### Sample Run

```

TYPE A MESSAGE OF UP TO 85 CHARACTERS ?
ABCDEFGHIJKLMNOPQRSTUVWXYZ
ZYXWVUTSRQPONMLKJIHGFEDCBA

```

## High Ball

No excitement at the office water cooler? Bored in study hall? Nothing to do on a rainy Saturday at home? Here's a fast game of Highball in which the highest-scoring player wins.

The game is built for five players. You type in the first name, or initials or other identifying label, for each of the five players. The computer draws straws and randomly assigns numbers to each player. It sorts to find which player has the *highest* number and declares that player the winner.

The number of players is controlled by the FOR/NEXT loop in lines 105 to 180. Change the number 5 at the end of line 105 to change the number of players.

This is an ideal bit of speedy entertainment. The game runs endlessly until you press BREAK to stop it.

### Program Listing

```

10 CLS: CLEAR: RV$=CHR$(27)+"p": NV$=CHR$(27)+"q"
20 DIM P$(5), S(5)
30 PRINT @ 56, RV$; "HIGH BALL"; NV$
40 LINE(95,7)-(149,15), 1, P
50 LINE(93,5)-(151,17), 1, B
100 PRINT @ 122, "TYPE IN FIRST NAMES OF FIVE PLAYERS"
105 FOR L=1 TO 5
110 PRINT @ 203, "PLAYER NUMBER "; L;
115 INPUT P$(L)
120 PRINT @ 219, SPACE$(20)

```

```

130 IF LEN(P$(L))>12 THEN PRINT @ 243,
    "NAME TOO LONG, TRY AGAIN":GOTO 110
140 PRINT @ 243,SPACE$(30)
150 FOR Q=1 TO VAL(RIGHT$(TIME$,2))
    :S(L)=INT(100*RND(1)):NEXT Q
160 IF L=1 THEN H=S(L)
170 IF S(L)>H THEN H=S(L)
180 NEXT L
190 CLS:PRINT
200 FOR L=1 TO 5
210 IF S(L)=H THEN PRINT S(L),P$(L)
    ;"<-----WINNER"
220 IF S(L)<>H THEN PRINT S(L),P$(L)
230 NEXT L
240 PRINT"PRESS ANY KEY TO PLAY AGAIN"
250 IF INKEY$="" THEN 250
270 GOTO 10

```

## Remarkable Person

This is a fun program to impress your friends with the speed and "intelligence" of your computer. It is a great demonstrator of computer graphics as shown on the computer's display. Note the reverse box!

Press any key on the keyboard to do a new name. Press BREAK to end the run.

### Program Listing

```

10 CLS: CLEAR
30 INPUT"WHAT IS YOUR FIRST NAME":FN$
40 INPUT"WHAT IS YOUR LAST NAME ":LN$
50 IF FN$="" OR LN$="" THEN 10
60 CLS
70 LF=LEN(FN$):LS=LEN(LN$)
80 BF=(21-LF-LS)/2
90 BS=BF
100 IF INT((LF+LS)/2)<>(LF+LS)/2 THEN
    BS=BS-1
110 IF 21-LF-LS<1 THEN PRINT "NAME TOO
    LONG",, "GIVE ME A SHORTER ONE"
    :GOTO 30

```

```

120 BF$=SPACE$(BF)
130 BS$=SPACE$(BS)
140 RV$=CHR$(27)+"p";NV$=CHR$(27)+"q"
150 LINE(59,15)-(185,40),1,B
160 LINE(57,13)-(187,42),1,B
170 PRINT @ 90,RV$;BF$;FN$;" ";LN$;BS$;NV$
180 PRINT @ 130,RV$"          IS          "
      ;NV$
190 PRINT @ 170,RV$;" A REMARKABLE PERSON
      ";NV$
200 IF INKEY$="" THEN 200
210 GOTO 10

```

## Computer Guesses Your Secret Number

Wow! Everybody has heard of the high/low game in which a computer selects a secret number and you try to guess it. But, what if *you* think up the secret number and the computer tries to guess it?

When you run this program, think of a secret number from one to 63. The computer will ask some easy questions and then guess your secret number every time!

To ask its questions, the computer displays sets of various numbers. It does this six times. Each time you tell it nothing more than whether or not your secret number is in the set of numbers shown on the display. You tell it this simply by pressing Y for yes or N for no.

After its sixth query, it will correctly tell you what the mystery number was. Naturally, you can't change the mystery number after the start of the game and you must tell the computer truthful answers to each of the six questions.

At the start of a RUN, the computer will ask for your name. You give it and then read an opening explanation billboard on the display screen. Then press ENTER to start. At completion of a game, press ENTER to play a new game. The program runs on endlessly until you stop it by pressing the BREAK key.



## Program Listing

```
10 CLS: CLEAR
20 INPUT "WHAT IS YOUR NAME"; N$
30 PRINT "OKAY, "; N$; ", THINK OF A NUMBER"
40 PRINT "FROM 1 TO 63"
45 INPUT "PRESS ENTER TO START"; Q$
50 CLS
55 S=0
60 FOR J=1 TO 6
70 T=1
75 PRINT;
80 FOR K=1 TO 32
90 READ A
100 IF A<10 THEN PRINT A; " ";:GOTO 120
110 PRINT A; " ";
120 IF K=8*T THEN PRINT;:T=T+1
130 NEXT K
140 PRINT N$; ", ";
150 INPUT "IS YOUR NUMBER HERE"; R$
160 IF LEFT$(R$,1)="Y" THEN S=S+2^(J-1)
170 IF LEFT$(R$,1)<>"Y" AND LEFT$(R$,1)
    <>"N" THEN 150
175 CLS
180 R$=" "
190 NEXT J
200 PRINT @ 90, "YOUR NUMBER IS"; S
210 PRINT @ 164, " ";: INPUT "TO PLAY
    AGAIN, PRESS ENTER"; P$
220 RESTORE
230 GOTO 10
300 DATA 1,3,5,7,9,11,13,15,17,19,21,23
    ,25,27,29,31,33,35,37
310 DATA 39,41,43,45,47,49,51,53,55,57
    ,59,61,63
320 DATA 2,3,6,7,10,11,14,15,18,19,22
    ,23,26,27,30,31,34,35,38
330 DATA 39,42,43,46,47,50,51,54,55,58
    ,59,62,63
340 DATA 4,5,6,7,12,13,14,15,20,21,22
    ,23,28,29,30,31,36,37,38
```

```

350 DATA 39,44,45,46,47,52,53,54,55,60
    ,61,62,63
360 DATA 8,9,10,11,12,13,14,15,24,25,26
    ,27,28,29,30,31,40,41,42,43,44,45
    ,46,47,56,57,58,59,60,61,62,63
370 DATA 16,17,18,19,20,21,22,23,24,25
    ,26,27,28,29,30,31,48,49
380 DATA 50,51,52,53,54,55,56,57,58,59
    ,60,61,62,63
390 DATA 32,33,34,35,36,37,38,39,40,41
    ,42,43,44,45,46,47,48,49
400 DATA 50,51,52,53,54,55,56,57,58,59
    ,60,61,62,63

```

### Sample Run

WHAT IS YOUR NAME? HELEN  
 OKAY, HELEN, THINK OF A NUMBER  
 FROM 1 TO 63  
 PRESS ENTER TO START?

1	3	5	7	9	11	13	15
17	19	21	23	25	27	29	31
33	35	37	39	41	43	45	47
49	51	53	55	57	59	61	63

HELEN, IS YOUR NUMBER HERE? Y

2	3	6	7	10	11	14	15
18	19	22	23	26	27	30	31
34	35	38	39	42	43	46	47
50	51	54	55	58	59	62	63

HELEN, IS YOUR NUMBER HERE? Y

32	33	34	35	36	37	38	39
40	41	42	43	44	45	46	47
48	49	50	51	52	53	54	55
56	57	58	59	60	61	62	63

HELEN, IS YOUR NUMBER HERE? N

YOUR NUMBER IS 11

# Draw Sketches

Create your own computer art! This program allows you to draw sketches directly on the display panel.

When you start a RUN, you will be presented with a menu offering a choice of instructions or drawing. Press I to read the instructions or D to draw. If I is pressed, you will find that drawing starts at the center of the display.

Press the U letter key to move the dot up. Press D to move down. Press L to move left and R to move right. Press Q anytime to quit drawing.

You may choose to draw either at the initial menu or after reading the instructions. At the initial menu, press D to draw. If you read instructions first, then press ENTER to set the dot at the center of the screen. After you hear five beeps, start drawing. Pressing Q at anytime causes the RUN to end. When the run ends you will leave BASIC and go back out to the Model 100 main program-directory menu.

## Program Listing

```
10 CLS
20 PRINT @ 133,"DRAW A SKETCH"
30 PRINT @ 211,"TO DRAW    PRESS D"
40 PRINT @ 246,"INSTRUCTIONS    PRESS I"
50 KA$=INKEY$
60 IF KA$="" THEN 50
70 IF KA$="D" THEN 100
80 IF KA$="I" THEN 500
90 GOTO 50
100 CLS
110 FOR L=1 TO 5
120 LINE(118,30)-(122,34),1,BF
130 FOR T=1 TO 50:NEXT T
140 CLS
150 BEEP
160 FOR T=1 TO 50:NEXT T
170 NEXT L
180 X=120:Y=32
190 GOTO 300
200 KB$=INKEY$
210 IF KB$="" THEN 200
220 IF KB$="Q" THEN MENU
```

```

230 IF KB$="U" THEN Y=Y-1
235 IF Y<0 THEN Y=0
240 IF KB$="D" THEN Y=Y+1
245 IF Y>63 THEN Y=63
250 IF KB$="L" THEN X=X-1
255 IF X<0 THEN X=0
260 IF KB$="R" THEN X=X+1
265 IF X>239 THEN X=239
270 IF KB$="U" OR KB$="D" OR KB$="L" OR
    KB$="R" THEN 300
280 GOTO 200
300 PSET(X,Y)
310 KB$=" "
320 GOTO 200
500 CLS
510 PRINT TAB(12);"INSTRUCTIONS"
520 PRINT"1)  DRAWING STARTS AT SCREEN
    CENTER"
530 PRINT"2)  PRESS U TO MOVE DOT UP,
    D TO MOVE"
540 PRINT"    DOWN, L TO MOVE LEFT, R
    MOVE RIGHT"
550 PRINT"3)  PRESS Q ANYTIME TO QUIT
    DRAWING"
560 PRINT @ 248,"NOW PRESS ENTER TO DRAW"
570 KC$=INKEY$
580 IF KC$=" " THEN 570
590 IF KC$="Q" THEN MENU
600 IF ASC(KC$)=13 THEN 100
610 GOTO 570

```

### Sample Run

```

          DRAW A SKETCH
          TO DRAW    PRESS D
INSTRUCTIONS    PRESS I
          INSTRUCTIONS
1)  DRAWING STARTS AT SCREEN CENTER
2)  PRESS U TO MOVE DOT UP, D TO MOVE
    DOWN, L TO MOVE LEFT, R MOVE RIGHT
3)  PRESS Q ANYTIME TO QUIT DRAWING

          NOW PRESS ENTER TO DRAW

```

## Dot Sweeper

This program endlessly prints LCD dots across and down, from upper left to lower right, on the display panel. After arriving at the lower right, it will start erasing at the upper left and erase down to the lower right.

A great way to make sure your entire LCD panel is working properly! Press BREAK to quit.

### Program Listing

```
10 CLS: CLEAR
100 FOR Y=0 TO 63 STEP 2
110 FOR X=0 TO 239 STEP 2
120 PSET(X,Y)
130 NEXT X
140 NEXT Y
200 FOR Y=0 TO 63 STEP 2
210 FOR X=0 TO 239 STEP 2
220 PRESET(X,Y)
230 NEXT X
240 NEXT Y
300 GOTO 100
```

## Stock Price Plotter

Graph the movement in price of your favorite equity! This program allows you to look at the change in stock price over 200 consecutive days.

The range of price entry is zero to 63. The stock symbol can be any length up to five characters. Enter prices for any number of days up to 200 days.

One especially nice — and unusual — feature of this program is its built-in self-test mode. Type the word TEST at the stock-symbol request and the machine will automatically run through a test pattern for you to see how it works!

By the way, it is not necessary for you to see the title and data if you want to eliminate a step. Press ENTER and see a blank graph grid.

On the graph, the Y axis is the price and the X axis is time in days, chronologically left to right. Use only as many

days as you wish, up to 200. Press ENTER with no data to exit the entry routine and get on with plotting the graph. Press any key after the graph is drawn to do a new one.

### Program Listing

```
10 CLS: CLEAR
20 DIM P(239)
30 PRINT @ 50, "STOCK PRICE PLOTTER"
40 LINE(58,6)-(174,16),1,E
50 PRINT @ 120, "SYMBOL OF STOCK TO PLOT
   : " ; : INPUT S$
60 SL=LEN(S$)
70 FOR Z=120 TO 159: PRINT CHR$(127);
   : NEXT Z
80 IF S$="TEST" THEN 500
100 FOR L=40 TO 239
110 PRINT @ 120, "PRICE NO." ; L-39;
   : INPUT P$
120 IF P$="" THEN 200
130 IF ASC(P$)<48 OR ASC(P$)>57 THEN 110
140 P(L)=VAL(P$)
150 T=L
160 P$=""
170 FOR Z=120 TO 159: PRINT CHR$(127);
   : NEXT Z
180 NEXT L
200 CLS
210 PRINT @ 121, S$
220 LINE(39,63)-(239,63)
230 LINE(39,0)-(39,63)
300 FOR L=40 TO T
310 PSET (L,63-P(L))
320 NEXT L
400 IF INKEY$="" THEN 400
410 GOTO 10
500 PRINT @ 125, "PLEASE WAIT FOR TEST
   PLOTTING"
510 FOR L= 40 TO 239
520 P(L)=30+17*(COS((L-39)*0.114))
530 Y=30-INT((L-39)/7)
540 PRINT @ 213, Y; "SECONDS "
```

```
550 NEXT L
560 T=239
570 GOTO 200
```

## Sine Wave

Here's another handsome LCD demonstrator, showing the intricate designs possible on the display panel.

The program has the computer place a SINE WAVE label at the top of the screen. Then it blacks out a large block of the screen as a solid reverse. It draws, from left to right, a simulation of a sine wave curve, top-to-bottom, top-to-bottom. A bisecting straight line is drawn through the middle of the display. And finally the picture is frozen, awaiting the press of any key. A keypress will blank the screen and restart the entire process. The program runs endlessly until you stop it by pressing the BREAK key.

### Program Listing

```
10 CLS
20 LINE(10,10)-(230,50),1,BF
30 PRINT @ 16,"SINE WAVE"
100 FOR S=10 TO 230
110 PRESET(S,30+17*(COS(S*.114)))
120 NEXT S
200 FOR X=10 TO 230
210 PRESET(X,30)
220 NEXT X
300 IF INKEY$="" THEN 300
310 GOTO 10
```

## Annual Sales Bar Graph

Plot sales results for six years. When you start a RUN, the computer will ask for total annual sales for each of six years. We have selected the years 1981 through 1986 to show how it works but you can change to whatever years you like.

Key in a dollar-value sales-result number for each year and press ENTER. After the sixth year's data has been



entered, the computer automatically switches to display and shows a bar graph for those six years of data.

Data numbers have to be in the range of zero to 33 so you may have to imagine some zeros or otherwise make your sales-results data conform to the zero to 33 entry range.

The graph display will hold, awaiting the press of any key to start over. When starting over, you may do entirely new and different numbers if you wish. The program runs endlessly until you press the BREAK key.

### Program Listing

```
10 CLS: CLEAR
20 DIM D(6)
100 PRINT "ENTER YEAR DATA OF 0-33"
110 FOR Y=1 TO 6
120 Y$="198"+RIGHT$(STR$(Y),1)
130 PRINT Y$;
140 INPUT D
150 D(Y)=D
160 D=0
170 NEXT Y
200 CLS
210 PRINT @ 13, "ANNUAL SALES"
220 PRINT @ 40, "1981"
230 PRINT @ 80, "1982"
240 PRINT @ 120, "1983"
250 PRINT @ 160, "1984"
260 PRINT @ 200, "1985"
270 PRINT @ 240, "1986"
300 FOR Y=1 TO 6
310 PRINT @ ((Y*40)+4), CHR$(224);
320 IF D(Y)=0 THEN 370
330 FOR P=1 TO D(Y)
340 PRINT CHR$(239);
350 NEXT P
360 PRINT CHR$(224)
370 NEXT Y
400 IF INKEY$="" THEN 400
410 GOTO 10
```

## Time Plot

Visualize time as advancing chronologically, left to right. You may plot up to 240 individual pieces of data in the DATA lines starting at line 300 in the program. Add DATA lines as needed.

Limit the values of your data to a range of +30 to -30. We have supplied sample data points in lines 300 to 410. You may eliminate those and supply your own numbers, each ranging from +30 to -30. You may include up to 240 items of data.

Upon starting a RUN, the screen displays and you press any key to get a data read and plot. The display is frozen so you can study it. Press any key to end the run.

### Program Listing

```
10 CLS
20 PRINT TAB(15); "TIME PLOT"
30 LINE(0,32)-(239,32),1
40 PRINT @ 254, "1-->240 MINS"
50 IF INKEY$="" THEN 50
100 FOR X=0 TO 239
105 ON ERROR GOTO 200
110 READ Y
120 PSET (X,32-Y)
140 NEXT X
200 PRINT @ 254, "OUT OF DATA "
210 IF INKEY$="" THEN 210
220 CLS
230 PRINT @ 254, "END OF RUN"
240 PRINT:PRINT:END
300 DATA -15,-13,-11,-9,-7,-5,-3,-1
310 DATA 1,3,5,7,9,11,13,15,14,12,10
320 DATA 8,6,4,2,0,-2,-4,-6,-8,-10,-12
330 DATA -14,-13,-11,-9,-7,-5,-3,-1,1
340 DATA 3,5,7,9,11,13,15,14,12,10,8
350 DATA 6,4,2,0,-2,-4,-6,-8,-10,-12
360 DATA -14,-15,-13,-11,-9,-7,-5,-3
370 DATA -1,1,3,5,7,9,11,13,15,14,12
380 DATA 10,8,6,4,2,0,-2,-4,-6,-8,-10
390 DATA -12,-14,-15,-13,-11,-9,-7,-5
400 DATA -3,-1,1,3,5,7,9,11,13,15,14
410 DATA 12,10,8,6,4,2,0,-2,-4,-6,-8
```

## Teacher's Grade-Curve Graph

The computer asks the teacher for the quantity of grades in each of the letters, A, B, C, D and F. Then a vertical bar graph is created on the display screen.

The range of quantities you can use for each letter grade must be from zero to 50.

After the graph is drawn, press any key on the computer to do another set. Press BREAK at anytime to end the run.

### Program Listing

```
10 CLS: CLEAR
20 GOSUB 1000
30 PRINT "TYPE IN THE TOTAL NUMBER FOR
  EACH GRADE"
40 PRINT "A, B, C, D, F      0 TO 50 PER
  LETTER GRADE"
100 GOSUB 1100
110 INPUT "NUMBER OF A's" ; A
120 GOSUB 1100
130 INPUT "NUMBER OF B's" ; B
140 GOSUB 1100
150 INPUT "NUMBER OF C's" ; C
160 GOSUB 1100
170 INPUT "NUMBER OF D's" ; D
180 GOSUB 1100
190 INPUT "NUMBER OF F's" ; F
200 CLS
210 PRINT @ 160, CHR$(147); " ";
220 GOSUB 1010
230 LINE(95,1)-(95,62)
240 LINE -(163,62)
245 LINE -(163,1)
250 PRINT @ 17, "A B C D F"
260 LINE(102,62)-(108,62-A), 1, BF
270 LINE(114,62)-(120,62-B), 1, BF
280 LINE(126,62)-(132,62-C), 1, BF
290 LINE(138,62)-(144,62-D), 1, BF
300 LINE(150,62)-(156,62-F), 1, BF
400 IF INKEY#="" THEN 400
410 GOTO 10
```

```

990 END
1000 PRINT SPACE$(14);
1010 PRINT "GRADE CURVE"
1020 RETURN
1100 PRINT @ 226, "  "
1110 PRINT @ 210, CHR$(154);
1120 RETURN

```

## Draw Geometric Shapes

Draw circles, triangles, squares, rectangles on the Model 100 Portable Computer LCD display screen.

Circle radius may be from 2 points to 31 points on the screen. The triangle base may be from 2 to 238 while its height may be from one to 63.

The side length of a square may be from one to 63. The square may be solid or outline. Press S for solid, a reverse filled-in black block. Or press L for an outline square.

Rectangle width may be from one to 239 and height on the screen from one to 63 points. Press S for a solid filled-in rectangle or L for an outline rectangle display.

After the object you select is drawn, press any key to do another one. The program runs until you press the BREAK key.

### Program Listing

```

10 CLS: CLEAR
20 PRINT "WHICH SHAPE DO YOU WANT TO DRAW:"
30 PRINT "CIRCLE, TRIANGLE, SQUARE,
   RECTANGLE"
40 PRINT
50 PRINT "PRESS C, T, S OR R"
60 KY$=INKEY$
70 IF KY$="" THEN 60
80 IF KY$="C" THEN 200
90 IF KY$="T" THEN 400
100 IF KY$="S" THEN 600
110 IF KY$="R" THEN 800
120 GOTO 60

```

```

200 CLS
210 PRINT"DIMENSIONS ARE NEEDED FOR A
    CIRCLE"
220 PRINT"RADIUS MAY BE 2 TO 31"
230 INPUT"RADIUS";D
240 IF D<2 OR D>31 THEN 230
250 CLS
260 FOR M=1 TO 22
270 X=120+D*SIN(C)
280 Y=32+D*COS(C)
290 PSET(X,Y)
300 C=C+6
310 NEXT M
320 GOTO 1000
400 CLS
410 PRINT"DIMENSIONS ARE NEEDED FOR A
    TRIANGLE"
420 PRINT"THE BASE CAN BE 2 TO 238"
430 INPUT"BASE";B
440 IF B<2 OR B>238 THEN 430
450 PRINT"THE HEIGHTH CAN BE 1 TO 63"
460 INPUT"HEIGHTH";H
470 IF H<1 OR H>63 THEN 460
480 BB=B/2:CLS
490 LINE(120,63)-(120+BB,63)
500 LINE(120+BB,63)-(120,63-H)
510 LINE(120,63-H)-(120-BB,63)
520 LINE(120-BB,63)-(120,63)
530 GOTO 1000
600 CLS
610 PRINT"DIMENSIONS ARE NEEDED FOR A
    SQUARE"
620 PRINT"EACH SIDE CAN BE 1 TO 63"
630 INPUT"LENGTH OF SIDE";L
640 IF L<1 OR L>63 THEN 630
650 PRINT"OUTLINE BOX OR FILLED IN SOLID?"
660 PRINT"PRESS L (OUTLINE) OR S (SOLID)"
670 K$=INPUT$(1)
680 IF K$="L" THEN 710
690 IF K$="S" THEN 740
700 GOTO 670

```

```

710 CLS
720 LINE(90,63)-(90+L,63-L),1,B
730 GOTO 1000
740 CLS
750 LINE(90,63)-(90+L,63-L),1,BF
760 GOTO 1000
800 CLS
810 PRINT"DIMENSIONS ARE NEEDED FOR A
      RECTANGLE"
820 PRINT"THE WIDTH CAN BE 1 TO 239"
830 INPUT"WIDTH";W
840 IF W<1 OR W>239 THEN 830
850 PRINT"THE HEIGHTH CAN BE 1 TO 63"
860 INPUT"HEIGHTH";H
870 IF H<1 OR H>63 THEN 860
880 PRINT"OUTLINE BOX OR FILLED IN SOLID?"
890 PRINT"PRESS L (OUTLINE) OR S (SOLID)"
900 K$=INPUT$(1)
910 IF K$="L" THEN 950
920 IF K$="S" THEN 980
930 GOTO 900
950 CLS
960 LINE(0,63)-(0+W,63-H),1,B
970 GOTO 1000
980 CLS
990 LINE(0,63)-(0+W,63-H),1,BF
1000 IF INKEY$="" THEN 1000
1010 GOTO 10

```

## People Graph

This unusual bar graph uses small figures of people to create the lines across the display.

Assume you are plotting total numbers of persons doing something per year. The computer will ask you for that total number of persons for each of six years. The annual range can be from one to 30.

After the graph is drawn, the display will be frozen. The computer awaits the press of any key to do a new graph. The program runs until you press BREAK.

## Program Listing

```
10 CLS: CLEAR: DIM Y(6)
20 PRINT SPACE$(15); "BAR GRAPH"
30 PRINT "TYPE IN NUMBER OF PERSONS FOR
  SIX YEARS"
40 PRINT SPACE$(8); "ONE TO 30 FOR EACH
  YEAR"
50 PRINT
60 FOR N=1 TO 6
70 PRINT "NUMBER PERSONS YEAR "; N;
80 INPUT Y(N)
85 IF Y(N)>30 THEN 70
90 NEXT N
100 CLS
110 PRINT
120 FOR N=1 TO 6
130 PRINT "YEAR "; N; " ";
140 FOR L=1 TO Y(N)
150 PRINT CHR$(148);
160 NEXT L
170 PRINT
180 NEXT N
190 IF INKEY$="" THEN 190
200 GOTO 10
```

## Issues Histogram

This horizontal chart, or histogram, is used for comparisons. It gives a visual or graphic comparison which is easily read.

For demonstration purposes, we have selected issues upon which people have commented. The graph shows the relative strength of those issues. The sample data is imbedded in program lines 500-550. Note that each item in a DATA line is paired with an issue name followed by a strength number. Naturally, these are imaginary data provided only for purposes of example in this book.

If you change the data, note that issue names are limited to a maximum of eight characters. Use data mul-



tiples of six items per set as we have done here. That means if you add data, add it in sets of six issues.

During a run, press any key to get the next data screen histogram. The machine will display *no more data* when none is available. You may use as many combinations of 6-issue sets as you wish. Limit the horizontal bar length in data lines to a maximum of 185.

### Program Listing

```
10 CLS
15 PRINTSPACE$(6);"RELATIVE STRENGTH
  OF ISSUES"
20 FOR L=1 TO 6
30 READ B$,X2
40 PRINT B$
50 X1=50:Y1=L*8:X2=X2+50:Y2=(L*8)+6
60 LINE(X1,Y1)-(X2,Y2),1,BF
70 NEXT L
200 IF INKEY$="" THEN 200
210 ON ERROR GOTO 300
220 GOTO 10
300 CLS
310 PRINT @ 50,"NO MORE DATA"
320 END
500 DATA DEFENSE,141,POLITICS,99
510 DATA RACE,47,SEX,112,PRICES,185
520 DATA RELIGION,15,WEATHER,162
530 DATA SCHOOLS,56,DRUGS,134
540 DATA CABLE TV,33,ALCOHOL,89
550 DATA FOOD,185
```

### Horizontal Dot-Plot Graph

Plot horizontally across the computer screen. The Y values on the X-Y axis must be limited to -32 to +31.

Press any key after the graph is drawn to do another. The program runs continuously until you press the BREAK key.

The program draws a horizontal screen-center bar for reference.

```

10 CLS: CLEAR: DIM Y(240)
20 PRINT @ 7, "HORIZONTAL DOT-PLOT GRAPH"
30 LINE (34,0)-(34,9),1
40 LINE -(200,9),1: LINE -(200,0),1
50 PRINT @ 80, "PLEASE TYPE IN A GRAPH
  LABEL OF UP TO 20 CHARACTERS";
60 INPUT N$
70 LN=LEN(N$)
80 IF LN<1 OR LN>20 THEN PRINT @ 80,
  SPACE$(80): GOTO 50
90 NS=INT((40-LN)/2)
100 LS=6*(NS-1): LE=238-LS
110 CLS: PRINT @ NS, N$
120 LINE (LS,0)-(LS,9),1
130 IF LN/2=INT(LN/2) THEN LINE -(LE,9),1
  : LINE -(LE,0),1
140 IF LN/2<>INT(LN/2) THEN LINE -(LE-6
  ,9),1: LINE -(LE-6,0),1
200 PRINT @ 80, "ENTER Y VALUES RANGING
  -32 TO +31, THEN"
210 PRINT @ 120, "PRESS ENTER, WITHOUT
  DATA, TO SHOW GRAPH"
300 A=A+1
310 PRINT @ 200, "WHAT IS THE Y VALUE AT
  X"; A;: INPUT R$
320 IF R$="" THEN 400
330 Y(A)=VAL(R$)
340 IF Y(A)<-32 OR Y(A)>31 THEN PRINT
  @ 200, SPACE$(39): GOTO 310
350 IF A>238 THEN 400
360 PRINT @ 200, SPACE$(39)
370 R$=""
380 GOTO 300
400 PRINT @ 80, SPACE$(160)
410 FOR X=0 TO A
420 PSET(X,31-Y(X))
430 NEXT X
440 LINE (0,31)-(239,31),1
500 IF INKEY$="" THEN 500
510 GOTO 10

```

# **44 Programs for the TRS-80 Model 100 Portable Computer**

**by Jim Cole**

This is a book of 44 complete programs, ready for immediate use in the TRS-80 Portable Computer. Here are the practical, useful, efficient sets of software needed by businessmen, teachers, students, parents, professionals, scientists, doctors, lawyers, engineers, hobbyists and everyone wishing to put the Model 100 Portable Computer to worthwhile use.

These programs are ready to type in, ready to run. Included are many sets of software listings for business, personal improvement, personal finance, screen graphics, even challenging game programs for a fun break.

*For businesses*, the computer will compute ad costs, track inventory, analyze cash flow, make change, select daily codes, and much more. *In personal finance*, the computer will compute mortgage loans, help you plan installment-loan purchases, show how money grows and when deposits double, and much more.

*For your personal use*, the computer will log your jogs, convert your time to universal, decipher Centigrade and Kelvin temperatures, keep a list of names and notes, teach you Morse code, work as an alarm clock, convert your numbers to binary, set and break some very, very secret codes, and even tell you what day of the year it is.

*Graphically*, it draws sketches on the LCD screen, sweeps dots, plots stock market prices, displays annual sales, plots time, curves teacher's grades, draws geometric shapes and histograms, and much more.

All of these and more are in this handy volume of exciting new and different software programs for the Model 100 and NEC PC-8200 series portable computers. These programs will run, exactly as they are here, in any computer using the same Microsoft version of BASIC as found in the TRS-80 Model 100. With slight modification to program lines, these programs will run on any BASIC computer. Try 'em all. You'll love 'em.

**ARCsoft Publishers**  
**Woodsboro, Maryland**